

Modelling and verifying the PIP protokernel in a deep embedding of C

Mohamed Sami Cherif

Supervised by

Paolo Torrini

Internship period

May 22 - August 11, 2017

EXTRA SMALL EXTRA SAFE

CRIStAL
Centre de Recherche en Informatique, Signal et Automatique de Lille

Université de Lille 1 SCIENCES ET TECHNOLOGIES

UFR IEEA
Formations en Informatique de Lille 1

Abstract

Formal proofs on program correctness are becoming longer and more complex especially when dealing with critical software. Therefore, it is essential to make these proofs as structured as possible, legible and amendable. The purpose of this research is to study proofs specified in DEC, an intermediate deeply embedded language for the translation of the PIP protokernel, a minimal OS kernel with provable isolation, to C. We specifically want to check whether the strict structuring of DEC is reflected in invariant proofs and to compare them to those in the shallow embedding in which PIP is specified. To that end, three distinct invariants in the shallow embedding were replicated in DEC. Their program functions were modelled modularly and the invariant proofs were done correspondingly.

Keywords : formal proofs, deeply embedded language, shallow embedding

Acknowledgement

First and foremost, I wish to thank the whole 2XS team of the CRIStAL laboratory for their friendly welcome especially Prof. *Gilles Grimaud*, head of the team, for allowing me to perform this internship which turned out to be such an amazing experience and my supervisor, Dr. *Paolo Torrini*, postdoctoral researcher, without whom this work would have never been successfully accomplished, for his precious assistance, his dedicated involvement in every step throughout the whole process, his precious advices and his encouragement.

Furthermore, I would like to specially thank Dr. *David Nowak* and Dr. *Vlad Rusu*, research scientists in the 2XS team, for their precious help and last but not the least Mrs. *Narjes Jomaa*, PhD student, for her detailed explanations and her valuable support.

Finally, I would like to express my special appreciation and thanks to Prof. *Francesco De Comite*, associate professor in charge of training periods for 3rd year students in Computer Science in Lille 1 university, for his availability and understanding.

Contents

List of Scripts

List of Figures

Acronyms

API Application Programming Interface

DSL Domain Specific Language

HAL Hardware Abstraction Layer

IAL Interrupt Abstraction Layer

IPC Inter-Process Communication

MAL Memory Abstraction Layer

MMU Memory Management Unit

OS Operating System

SOS Structural Operational Semantics

TCB Trusted Computing Base

1. Introduction

The following report describes the activities carried out during a 12-week, full-time internship at the Research Center in Computer Science, Signal and Automatic Control of Lille (CRIStAL) [2]. This internship revolves around the PIP pretokernel and the DEC language currently being developed by the 2XS team. PIP [13] is an OS protokernel specified in Coq in terms of a shallow embedding of a fragment of the C language on which memory isolation is being proved using Hoare logic. DEC [16] is a language, deeply embedded in Coq, designed as an intermediate language for the translation of PIP to C. Specifying programs in the deep embedding has the great advantage of simplifying their syntactic manipulation. It also ensures a stricter structuring of program expressions. **Therefore, we want to see whether this structure is reflected in the proofs and compare such proofs to those already given in the shallow embedding.**

To that end, we will consider three distinct invariants in the shallow embedding. For each of them, we will model the program function in the deep embedding and carry out the proof for its corresponding invariant. The function of the first invariant reads the memory. The second one writes in the memory. The last one is a recursive function. One of the invariants propagates all of PIP's properties which include memory isolation, vertical sharing, kernel data isolation and consistency. We followed a modular approach in our implementation and we engineered our proofs correspondingly. We also did some preliminary work mostly to get familiar with Hoare logic and the deep embedding.

The first part of the report offers an overview of the CRIStAL laboratory, the 2XS team and their research activities. The second part is dedicated to the background work and focuses on the PIP protokenerel, its proof oriented design, its properties, its data structures, Hoare logic theory as well as the deep embedding and its constructs. The last part is dedicated to our contributions, detailing how we modelled the required functions, how we engineered our proofs in the deep embedding and our observations about the comparison between the deep and shallow proofs.

2. CRIStAL laboratory

2.1 History

CRIStAL[1] [2] (Research Center in Computer Science, Signal and Automatic Control of Lille), founded on the 1st of January 2015, is a laboratory of CNRS[2] (National Center for Scientific Research), Lille 1 university and Centrale Lille in partnership with Lille 3 University, Inria (French National Institute for computer science and applied mathematics) and Mines Telecom Institute. It is the result of the fusion of LAGIS[4] (Laboratory of Automatic Control, Computer Engineering and Signal) and LIFL[4] (Laboratory of Fundamental Computing of Lille) to federate their complementary competencies in information sciences. It is a member the interdisciplinary research institute IRCICA[5] (Research Institute on Software and Hardware Components for Advanced Information and Communication in Lille).

CRIStAL is located in Villeneuve d'Ascq city in Lille, the capital of the Hauts-de-France region and the prefecture of the Nord department and the fourth largest urban area in France after Paris, Lyon and Marseille. Chaired by *Prof.Olivier Colot*, it harbours about 430 personnel with exactly 228 permanents and more than 200 non-permanents. Permanent researchers are divided among more than 30 teams working on different themes and projects.

Figure 2.1: CRIStAL laboratory Location

[1] "Centre de Recherche en Informatique, Signal et Automatique de Lille"
[2] "Centre national de la recherche scientifique"
[3] "Laboratoire d'Automatique, Génie Informatique et Signal"
[4] "Laboratoire d'Informatique Fondamentale de Lille"
[5] "Institut de recherche sur les composants logiciels et matériels pour l'information et la communication avancée de Lille"

2.2 Research activities

CRIStAL's research activities concern topics related to the major scientific and societal issues of the moment such as: Big Data, software, computer imaging, human-machine interactions, robotics, control and supervision of large systems, intelligent embedded systems, bioinformatics... The laboratory is involved in the development of revolutionary platforms such as Pharo, a pure object oriented language and a powerful yet simple development environment used worldwide.

2.3 2XS team

The 2XS (eXtra Small, eXtra Safe) team is working on highly constrained embedded devices, precisely on designing software and hardware that are secure, safe and efficient. Research in this team is focused on defining new system architectures or new languages to allow fast development of reliable embedded software. The team addresses issues concerning memory footprint, energy consumption and security and takes profit from proficiencies in formal verification, hardware/software co-design and operating system architectures to tackle the aforementioned issues.

The team is led by *Prof. Gilles Grimaud* and has 15 members[1] as well as several trainees and most of its current work mainly revolves around the PIP project and its applications.

Figure 2.2: The 2XS team organizational chart [2]

[1] as of August 24, 2017

3. Background : PIP project

3.1 The PIP protokernel

3.1.1 A minimal OS kernel with provable isolation

An operating system is organized as a hierarchy of layers [12], each one constructed upon the one below. Each layer focuses on an essential role of the operating system such as memory management, multiprogramming and input/output. Generally speaking, while developing a kernel for an operating system based on the layered approach, the designers have a choice where to draw the kernel-user boundary. Traditionally, all the layers went in the kernel, but that is not necessary. In fact, putting as little as possible in kernel mode is safer because kernel bugs can bring down the system instantly. In contrast, user processes/layers are set up to have less power so that a bug there may not be fatal.

Various studies on bug density, relatively to the developed module size, age as well as other factors, have been conducted (e.g. *Basilli* and *Perricone* in 1984; and *Ostrand* and *Weyuker* in 2002). A ballpark figure for serious industrial systems is ten bugs per a thousand lines of code. Operating systems are sufficiently buggy that computer manufacturers put reset buttons on them, something the manufacturers of cars, TV sets and stereos do not do, despite the large amount of software in these devices. Furthermore, Operating systems generally present hardware resources to applications through high-level abstractions such as (virtual) file systems.

Therefore, we can distinguish several OS kernel families such as :

- **Monolithic kernels :** All the OS layers are put in kernel space as in Windows and Linux;
- **Microkernels :** The basic idea behind a microkernel design is to achieve high reliability by splitting the operating system up into small, well-defined modules, only one of which, the microkernel, runs in kernel mode;
- **Exokernels :** The idea behind an exokernel is to force as few abstractions as possible on application developers, enabling them to make as many decisions as possible about hardware abstractions.

As a **minimal OS Kernel with provable isolation**, PIP [13] focuses more on security and safety without sacrificing efficiency and ensures memory isolation between different tasks running on the same device. PIP's algorithmic part is written in Gallina, the language of the Coq proof assistant, in a monadic style that allows direct translation into free standing C. We will refer to this implementation in *Gallina* as the shallow embedding in contrast to the deep embedding introduced in section 3.3 p.22.

Although the closest kernel design to PIP is the exokernel [8], PIP does not belong to any of the kernel families featured in the state of art, but it is the first member of a new kernel family, **protokernels**, as compared to most microkernels and exokernels, the TCB in PIP is even more restricted :

- Scheduling and IPC are done in user mode unlike a microkernel;
- Multiplexing is also done in user mode unlike an exokernel.

whereas the kernel mode is only for **multi-level MMU control and configuration** (virtual memory) and **context switching**. This not only ensures less bugs but also more feasibility of formal proof that will warrant the memory isolation property of the protokernel.



Figure 3.1: Software layers of an OS built on top of PIP [8]

3.1.2 Horizontal isolation & vertical sharing

PIP can be used to partition the available memory which will be initially allocated to the root partition on top of PIP. Any partition can read and write in the memory of its descendants. However, partitions in different branches of the partition tree are disjoint. The former is referred to as **vertical sharing** and the latter as **horizontal isolation** [7]. Needless to say, all memory lent to PIP for storing kernel data,such as when creating partitions, is inaccessible for all partitions to prevent messing up PIP data structures which means that the kernel data is totally isolated.



Figure 3.2: Horizontal isolation & vertical sharing in PIP [7]

Let us consider a more realistic partition tree, as shown in figure 3.1.2, in which we consider *Linux* and *FreeRTOS* as sub-partitions of a root partition, multiplexer. Knowing that *FreeRTOS* is a real-time OS that does not isolate its tasks, we have easily secured it with task isolation by porting it on PIP.



Figure 3.3: FreeRTOS task isolation using PIP [8]

3.1.3 Proof-oriented design

PIP's design

PIP's isolation properties are meant to be formally proven independently from the platform it's running onto. Consequently, the algorithm and the architecture dependant part were separated [7]. Indeed, as shown in figure 3.4, PIP is split into two distinct layers :

- **HAL :** gives direct access to the architecture and hardware;
- **API :** implements the algorithmic part to configure the virtual memory and the hardware.

The API code is written and proven using the Coq proof assistant, and uses the interface provided by the HAL to perform any hardware related operation. the proofs are based on Hoare logic theory introduced in section 3.2.2 p.17.



Figure 3.4: PIP design [7]

PIP's HAL is split into three components, as shown in figure 3.5, each handling a specific part of the target platform's hardware :

- **Memory Abstraction Layer (MAL) :** provides an interface for the configuration of the MMU chip ([13], MAL.v, MALInternal.v);
- **Interrupt Abstraction Layer (IAL) :** provides an interface to dispatch interrupts and configure hardware;
- **Bootstrap :** contains the low-level code required to boot the system.

PIP only provides system calls for management of the partitions and for context switching, thus reducing the TCB to its bare minimum as explained in section 3.1.1 p.4. This user exposed API ([13], Services.v) can be called by any partition using a platform dependant call :

- **createPartition :** creates a new child (sub-partition) into the current partition;
- **deletePartition :** removes a child partition and puts all its used pages back in the current partition;
- **prepare :** adds required configuration tables into a child partition to map a new virtual address;
- **collect :** removes the empty configuration tables which are not used anymore and gives it back to the current partition;
- **countToMap :** returns the amount of configuration tables needed to perform a mapping for a given virtual address;
- **addVaddr :** maps a virtual address into the given child;
- **removeVAddr :** removes a given mapping from a given child.

This API is sufficient as far as memory requirements are concerned but it lacks a way to handle interrupts. Hardware interrupts are implicitly handled by PIP and automatically dispatched to the root partition, while software interrupts, such as system calls, are notified to the parent partition of the caller [7] and can be managed by these two additional services of PIP :

- **dispatch :** notifies an interrupt to a given partition, interrupting its current control flow and backing it up for a further resume call;
- **resume :** restores a previously interrupted context.



Figure 3.5: HAL and API relationship [7]

Formal proofs using Coq



Coq [15] is the result of about 30 years of research. It started in 1984 as an implementation of the Calculus of Constructions, an expressive formal language, at INRIA by *Thierry Coquand* and *Gérard Huet* and was extended, later in 1991, by *Christine Paulin* to the Calculus of Inductive Constructions [9].

Coq is a **formal proof management system**. It provides a formal language to write mathematical definitions, executable algorithms and theorems together with an environment for semi-interactive development of machine-checked proofs. Typical applications include the certification of properties of programming languages (e.g. the *CompCert* compiler certification project, or the *Bedrock* verified low-level programming library), the formalization of mathematics and teaching. It implements a program specification and mathematical higher-level language called *Gallina* that is based on the Calculus of Inductive Constructions combining both a higher-order logic and a richly-typed functional programming language.

As a **proof development system**, Coq provides interactive proof methods, decision and semi-decision algorithms as well as a tactic language letting the user define his own proof methods. Furthermore, as a **platform for the formalization of mathematics or the development of programs**, Coq provides support for high-level notations, implicit contents and various other useful kinds of macros.

For this project, the latest version of Coq, 8.6 released in December 2016, was used. It features among other things a faster universe checker, asynchronous error processing, proof search improvements, generalized introduction patterns, a new warning system and subterm selection algorithm.

The Coq proof assistant provides us with powerful tactics to perform proofs [14]. Here is a non exhaustive list of these tactics :

- **simpl :** simplifies the current goal;
- **assumption :** solves the current goal if it is computationally equal to a hypothesis;
- **reflexivity :** solves the current goal if it is a valid equality;
- **unfold t :** unfolds the definition of t in the current goal;

- **clear H :** removes hypothesis H from the context;
- **auto :** tries to solve the current goal automatically by using a collection of tactics;
- **rewrite H :** uses an equality hypothesis H to replace a term in the current goal;
- **induction t:** applies the induction principle for the type of t, generates subgoals as many as there are constructors, and adds the inductive hypotheses in the contexts;
- **inversion H :** resembles the induction tactic, but pays attention to the particular form of the type of H, and will only consider the cases that could have been used, so it discards some impossible cases quickly and efficiently;
- **omega :** carries out an automatic decision procedure for *Presburger* arithmetic;
- **apply H :** mainly tries to unify the current goal with the conclusion of the type of H. If it succeeds, then the tactic returns as many subgoals as non-dependent premises of the type of H;
- **f_equal :** applies to a goal of the form $f\ a_1 \ldots a_n = g\ b_1 \ldots b_n$ and leads to subgoals $f = g$ and $a_1 = b_1$ and so on up to $a_n = b_n$. Amongst these subgoals, the simple ones are automatically solved;
- **contradiction :** tries to find a contradiction amongst the hypotheses.

Coq also provides several tactics to deal with higher-order logic, particularly targeting binary connectives and quantifiers such as **split** for conjunction, **left** and **right** for disjunction, **intros** for implication and universal quantifiers and **exists** for existential quantifiers when they are in the goal. In addition, We have **destruct** for conjunction and disjunction, **apply** for implication, **elim** for existential quantifiers and **specialize** for universal quantifiers when they are in the hypothesis. Furthermore, many tactics can be preceded with the letter *e* like **eauto**, **eassumption** and **eapply** to deal with existential variables. Some tactics can also be applied on hypotheses if we use the reserved clause *in* with the name of the hypothesis such as unfold, apply and simpl.

3.1.4 In-depth understanding of PIP's data structures

The memory

PIP uses several data structures per partition [7], which will represent the global state of the partition's memory state. This is necessary as it has to keep track of pages allocated to partitions in order to allow or deny derivation and partition creation while preserving the required properties.

Figure 3.6 shows a partition tree example consisting of a partition, Parent, that has a single child, Child1. A partition is identified by a partition descriptor which is a page number essential to access the whole data structure of a partition. In our case, the partition descriptors of Parent and Child1 are respectively 1 and 12. The pages lent to PIP to manage a partition are organized in a tree with three branches : the MMU tables, the first shadow and the second shadow. The aim of this organization is to keep additional information about each page lent to a partition. Moreover these structures have multiple goals :

- **Access control :** the first shadow is used to avoid deriving the same page multiple times;
- **Performance :** the second shadow and the configuration list are used to quickly find the virtual address of a page without having to parse the whole virtual space when the parent partition reclaims it.

As such, adding an indirection table in the MMU configuration requires two additional pages for the shadows. Therefore, this model is estimated to require roughly three times the amount of memory a simple virtual environment would need, nevertheless it provides a secure and an efficient API.

To prove the isolation properties on this memory structure, it is essential to ensure its consistency relative to the partition tree, the well-typedness as well as some other consistency properties that will be detailed in section 3.2.2 p.17.



Figure 3.6: An example of a partition tree [7]

The memory state

The PIP memory state [6] ([13], ADT.v, Hardware.v) is defined as follows :

Script 3.1: PIP state definition

```
Record state : Type :=
{ currentPartition: page ; memory: list (paddr * value) }.
```

The list in the state corresponds to the physical memory and maps physical addresses to values. A physical address is defined by a physical page number and a position into this page :

Script 3.2: paddr type definition

```
Definition paddr : Type := page * index.
```

where pages and indexes are positive integers bounded respectively by the overall number of pages and the table size :

Script 3.3: page & index type definitions

```
Record page := { p :> nat ; Hp : p < nbPage }.
Record index := { i :> nat ;  Hi : i < tableSize }.
```

Different types of values could be stored in the physical memory by Pip. So, the type *value* is an inductive type defined as follows :

Script 3.4: value type definition

```
Inductive value : Type :=
        | PE: Pentry → value
        | VE: Ventry → value
        | PP: page   → value
        | VA: vaddr  → value
        | I:  index  → value.
```

The type *Pentry*, which represents a physical entry, consists of a physical page number along with several flags :

Script 3.5: Pentry type definition

```
Record Pentry : Type := {
        read:    bool ;
        write:   bool ;
        exec:    bool ;
        present: bool ;
        user:    bool ;
        pa:      page }.
```

Finally, the *Ventry* type consists of a virtual address with a boolean flag :

Script 3.6: Ventry type definition

```
Record Ventry : Type := { pd: bool ; va: vaddr }.
```

with virtual addresses modelled as a list of indexes of length the number of levels of the MMU plus one :

Script 3.7: vaddr type definition

```
Record vaddr : Type :=
{ va :> list index ; Hva : length va = nbLevel + 1  }.
```

The PIP monad

The previous memory state definition is used to define a monad, called *LLI*, that performs the computation of PIP programs. When we execute a PIP program function starting from a certain memory state, we are not only interested in its resulting value but also in the resulting state. This is essential for stateful operations which modify the memory state. It is also important to deal with undefined behaviour. The *LLI* ensures all of these specifications and is defined as follows ([13],Hardware.v) :

Script 3.8: PIP LLI monad

```
Inductive result (A : Type) : Type :=
| val : A → result A
| undef : nat → state → result A.

Definition LLI (A :Type) : Type := state → result (A * state).
```

Using the LLI monad, some instructions were explicitly defined as follows ([13],Hardware.v) :

Script 3.9: Explicitly defined instructions in the shallow embedding

```
(*Return instruction*)
Definition ret {A : Type} (a : A) : LLI A :=
  fun s ⇒ val (a , s) .

(*Assignment instruction*)
Definition bind {A B : Type} (m : LLI A)
                (f : A → LLI B) : LLI B :=
  fun s ⇒ match m s with
    | val (a, s') ⇒ f a s'
    | undef a s' ⇒ undef a s'
    end.

(*Skip instruction*)
Definition put (s : state) : LLI unit :=
  fun _ ⇒ val (tt, s).

(*For undefined behaviour*)
Definition undefined {A : Type} (code : nat) : LLI A :=
  fun s ⇒ undef code s.
```

3.2 Hoare logic

3.2.1 Introduction to Hoare logic theory

How can we argue that a program is correct ? Nowadays, Building reliable software is becoming more and more difficult considering the growing scale, specifications and complexity of modern systems. Therefore, tests alone can no longer ascertain the reliability of programs especially if we're talking about critical systems. Logicians, computer scientists and software engineers have responded to these challenges by developing different kinds of techniques some of which are based on formal reasoning about properties of software and tools for helping validate these properties. One of these reasoning techniques that was used to prove PIP's properties is *Floyd–Hoare logic*, often shortened to just **Hoare Logic**. It was introduced in 1969 by the British computer scientist and logician Tony Hoare [4] and it continues to be the subject of intensive research right up to the present day. It is not only a natural way of **writing down specifications of programs** but also a technique for **proving that programs are correct** with respect to such specifications. A Coq implementation of Hoare logic on a simple imperative programming language called Imp is explained in *Software Foundations* [10].

Let S be a program that we want to execute starting from a certain state s, P a predicate on the state describing the condition S relies on for correct execution and Q a predicate on the resulting state after the execution of S describing the condition S establishes after correctly running. Knowing that P is verified on s, if we prove that Q is verified after the execution of s, we can ascertain that S is partially correct. And by partial correctness of S we mean that S is correct if it terminates. Using standard Hoare logic, only partial correctness can be proven, while termination needs to be proven separately. This triple S, P and Q, written as **{P} S {Q}** , is referred to as a **Hoare triple**. The assertions P and Q are respectively referred to as the **precondition** and the **postcondition**.

For example let's consider simple Hoare triples about an assignment command :

- $\{X = 2\}X := X + 1\{X = 3\}$: is a valid Hoare triple, that can be easily formally proved in Coq, since the postcondition is verified after the execution of the assign command relatively to the precondition;
- $\{X = 2\}X := X * 2\{X = 3\}$: is not a valid Hoare triple since the postcondition would not be verified after the execution of the command.

Now, let's introduce some facts and rules about Hoare triples :

1. If an assertion P implies another precondition P' of a valid Hoare triple {P'} S {Q} then {P} S {Q} is also a valid Hoare triple. This is referred to as **weakening the precondition of a Hoare triple** and can be formally defined as follows :

$$\frac{\text{P} \rightarrow \text{P'} \qquad \{\text{P'}\}\ \text{S}\ \{\text{Q}\}}{\{\text{P}\}\ \text{S}\ \{\text{Q}\}}\ \text{Hoare\_weaken}$$

2. If we can weaken a Hoare triple, we expect it to have a **weakest precondition**. This notion was introduced by Dijkstra in 1976 [3] and is very important since it enables us to prove total correctness and in particular program termination. Indeed, if a program doesn't terminate, its weakest precondition would be *True* and it would verify any postcondition.

3. If we consider a language containing a **SKIP instruction** which practically does nothing, we can affirm that this command preserves any property which means :

$$\frac{}{\{\text{P}\}\ \text{SKIP}\ \{\text{P}\}}\ \text{Hoare\_skip}$$

4. If we consider a language that allows **assignments**, in the form of $X := a$, then we can conclude that an arbitrary property Q holds after such assignment if we assume Q[X→a] which means Q with all occurrences of X replaced by a :

$$\frac{}{\{\text{Q}[\text{X}\rightarrow\text{a}]\}\ \text{X:= a}\ \{\text{Q}\}}\ \text{Hoare\_assign}$$

5. Generally speaking, every Program is built using **sequencing** of commands that we will write as C1;C2. Our aim is to prove a Hoare triple on this sequence with P and Q respectively as the precondition and the postcondition. This requires proving that C1 takes any state where P holds to a state where an intermediate assertion I holds and C2 takes any state where I holds to one where Q holds which could be formally stated as :

$$\frac{\{\text{P}\}\ \text{C1}\ \{\text{I}\} \qquad \{\text{I}\}\ \text{C2}\ \{\text{Q}\}}{\{\text{P}\}\ \text{C1;C2}\ \{\text{Q}\}}\ \text{Hoare\_seq}$$

6. Finally, let's not forget **conditional structures** that we can write in the form of *IF B THEN C1 ELSE C2*. To verify a Hoare triple about this instruction with P and Q respectively as the precondition and postcondition, we need to consider both cases where B is evaluated to *true* and *false* and prove that Q holds on the resulting state in each one. We can also disregard a case if there is some sort of contradiction relatively to the property P. This rule can be written as follows :

$$\frac{\{P \bigwedge B\}\ C1\ \{Q\} \qquad \{P \bigwedge \neg\ B\}\ C2\ \{Q\}}{\{P\}\ IF\ B\ THEN\ C1\ ELSE\ C2\ \{Q\}}\ \text{Hoare\_if}$$

The rules defined above are in their simplest form and we need to adapt them to the language or semantics we are working on. Also, This list is not exhaustive. For example, we did not define a rule for *While* loops because the Coq model of PIP doesn't actually use them and uses recursion instead. Likewise, We are sure that all PIP's functions terminate since recursive functions are bounded by a maximum number of iterations.

3.2.2 Hoare logic in the shallow embedding

The Hoare logic devised for the shallow embedding is slightly different from what we saw in the previous section [5]. A program in the shallow embedding is defined in a monadic style that returns a pair of values, the first being the resulting state and the second the value returned by the program which generally corresponds to a function. Thus, as shown in script 3.10 ([13], Hardware.v), **the postcondition is a predicate on the resulting state and the returned value** so that we can verify it and propagate it in the case of long proofs. The notation chosen for the Hoare triple stays the same as the one mentioned in the previous section except for the brackets that are doubled since single brackets are already used in Coq.

Script 3.10: Hoare triple in the shallow embedding

```
Definition hoareTriple {A : Type} (P : state → Prop)
(m : LLI A) (Q : A → state → Prop) : Prop :=
  ∀ s, P s → match (m s) with
    | val (a, s') => Q a s'
    | undef _ _=> False
    end.

Notation "{{ P }} m {{ Q }}" := (hoareTriple P m Q)
  (at level 90, format "'[' '[' {{  P  }}  ']' '/  '
  '[' m ']' '[' {{  Q  }} ']' ']'") : state_scope.
```

The weakening lemma, extensively used in PIP's proofs, was also defined and proven in the following script ([13], WeakestPreconditions.v) :

Script 3.11: Weakening Hoare triples in the shallow embedding

```
Lemma weaken (A : Type) (m : LLI A)
(P Q : state → Prop) (R : A → state → Prop) :
  {{ Q }} m {{ R }} →
  (∀ s, P s → Q s) →
  {{ P }} m {{ R }}.
Proof.
intros H1 H2 s H3.
case_eq (m s); [intros [a s'] H4 | intros a H4 ];
apply H2 in H3; apply H1 in H3;
try rewrite H4 in H3; trivial.
intros. rewrite H in H3. assumption.
Qed.
```

Some instructions were considered as primitive like conditional structures while some others were explicitly defined like assignments, in the form of ***perform x := m in e*** where the value of the program m gets assigned to x in the evaluation of the program e, defined in script 3.9 p.14. Furthermore, a Hoare logic rule was devised for assignments as follows ([13], WeakestPreconditions.v) :

Script 3.12: Hoare triples assignment rule in the shallow embedding

```
Lemma bind (A B : Type) (m : LLI A) (f : A → LLI B)
(P : state → Prop) ( Q : A → state → Prop)
(R : B → state → Prop) :
  (∀ a, {{ Q a }} f a {{ R }}) →
  {{ P }} m {{ Q }} →
  {{ P }} perform x := m in f x {{ R }}.
Proof.
intros H1 H2 s H3; unfold bind;
case_eq (m s); [intros [a s'] H4 | intros k s' H4];
apply H2 in H3; rewrite H4 in H3; trivial.
case_eq (f a s'); [intros [b s''] H5 | intros k s'' H5];
apply H1 in H3; rewrite H5 in H3; trivial.
Qed.
```

Finally, to reason with Hoare logic we need to formally specify the properties we want to prove [7], the most important being partition memory isolation, kernel data isolation, vertical sharing and consistency mentioned in 3.1.2 p.6. To that end, the following necessary functions on PIP's data structures were defined ([13], StateLib.v) :

- **getChildren :** returns the list of all children of a given parent partition in the partition tree of a given state;
- **getAncestors :** returns the list of all ancestors of a given partition in the partition tree of a given state;
- **getMappedPages :** returns the list of mapped pages of a given partition;
- **getAccessibleMappedPages :** returns the list of all mapped pages, of a given partition, marked as accessible;
- **getUsedPages :** returns the list of all the pages that are used by a given partition including the pages lent to PIP;
- **getConfigPages :** returns the list of configuration pages lent to PIP to manage a given partition;
- **getPartitions :** returns the list of all existing partitions of a given state which is naturally obtained by a search on the partition tree.

The first property defines horizontal isolation between children i.e., for any state s of the system, all memory pages owned by two different children of a given parent partition in the partition tree must be distinct. This property is formally defined in the Coq proof assistant as follows ([13], Isolation.v) :

Script 3.13: Horizontal isolation

```
Definition horizontalIsolation s :=
∀ parent child1 child2,
parent ∈ getPartitions s →
child1 ∈ getChildren parent s →
child2 ∈ getChildren parent s →
child1 ≠ child2 →
(getUsedPages child1 s) ∩ (getUsedPages child2 s) = ∅.
```

The second property defines vertical sharing between a parent partition and its children i.e. it ensures that all pages mapped by a child are mapped in its parent. This property is formally defined as follows ([13], Isolation.v) :

Script 3.14: Vertical sharing

```
Definition verticalSharing s :=
∀ parent child,
parent ∈ getPartitions s →
child ∈ getChildren parent s →
getUsedPages child s ⊆ getMappedPages parent s.
```

The third property defines kernel data isolation meaning it ensures that for any state of the system and any given two possibly-equal partitions, the code running in the second partition cannot access the pages containing configuration tables lent to PIP to manage the first partition. This property is formally defined as follows ([13], Isolation.v) :

Script 3.15: Kernel data isolation

```
Definition kernelDataIsolation s :=
∀ partition1 partition2,
partition1 ∈ getPartitions s →
partition2 ∈ getPartitions s →
(getAccessibleMappedPages partition1 s) ∩
(getConfigPages partition2 s) = ∅.
```

The last property of consistency is mandatory to prove the previously detailed properties . Consistency encompasses different sub-properties that were divided into four categories. As the definition of this property is quite cumbersome, we will only disclose these categories and illustrate each one with an example ([13], Consistency.v):

- **Partition tree structure :** the properties in this category ensure that the partition tree of a given state is consistent relatively to its awaited structure preset in section 3.1.4 p.11. For example, one of the properties defined in this category and called *noCycleInPartitionTree* ensures that there is no cycle in the partition tree of a given system i.e. each partition is different from all its ancestors in the partition tree. This property is formally defined as follows :

  Script 3.16: Example of partition-tree-consistency property

  ```
  Definition noCycleInPartitionTree s :=
  ∀ ancestor partition,
  partition ∈ getPartitions s →
  ancestor ∈ getAncestors partition s →
  ancestor ≠ partition.
  ```

- **Flag semantics :** This category focuses on the signification of flags used in the data structures. For instance, a property called *isPresentNotDefaultIff* states that the associated page number of a physical entry whose *present* flag is set to *false*, corresponds to the predefined default page value. This property is formally defined as shown in script 3.17 where *readPhyEntry* and *readPresent* are predefined functions that respectively read, for a given table, index and memory state, the present flag of a physical entry and its value :

Script 3.17: Example of a flags-semantics-consistency property

```
Definition isPresentNotDefaultIff s :=
∀ table idx,
readPresent table idx (s.memory) = Some false ↔
readPhyEntry table idx (s.memory) = Some defaultPage.
```

- **Pages properties :** This category concerns several properties about the partitions' mapped pages, the pages lent to the kernel as well as the relations between them. For example, the *noDupMappedPagesList* requires that the mapped pages of partitions be distinct from each other, using the function *NoDup* which verifies whether a list contains duplicates or not. This property is formally defined as follows :

Script 3.18: Example of a pages-consistency property

```
Definition noDupMappedPagesList s :=
∀ partition, partition ∈ getPartitions s →
NoDup (getMappedPages partition s).
```

- **Well-typedness :** This last category concerns kernel data types and ensures that PIP doesn't contain any type confusion. For instance, when PIP writes a virtual address in the memory, it should be read later as a virtual address which is ensured by a property called *dataStructurePdSh1Sh2asRoot*. Otherwise, it is considered as an undefined behaviour in the model.

The most difficult system call invariant, called *createPartition*, was successfully proven ([13], CreatePartition.v). It required about 60000 lines of code to prove. It is defined as follows :

Script 3.19: createPartition Hoare triple

```
Lemma createPartition (descChild pdChild
          shadow1 shadow2 list : vaddr) :
{{fun s ⇒ partitionsIsolation s ⋀ kernelDataIsolation s
          ⋀ verticalSharing s ⋀ consistency s }}
createPartition descChild pdChild shadow1 shadow2 list
{{fun _ s  ⇒ partitionsIsolation s ⋀ kernelDataIsolation
             s ⋀ verticalSharing s ⋀ consistency s }}.
```

3.3 The deep embedding

3.3.1 DEC

DEC [16] (Deep Embedding of a terminating C-style language with effects) is a Domain Specific Language (DSL) embedded in *Gallina* and developed as an intermediate language for the translation of PIP to C. DEC Relies on Coq modules to define parametrisable state and identifiers types which are both specified in a module type (*IdModType* shown in annex A p.60). The dynamic semantics of DEC is specified in the small-step style of Structural Operational Semantics (SOS), introduced by *Plotkin* [11], which formally describe how the individual steps of a computation take place enabling us to perform software verification by reasoning on formulas' structure. In small-step semantics, program behaviour is specified in terms of atomic steps and this makes it easier to specify constructs in dependently of one another [1].

The DEC language has the following properties ([16], README.txt) :

1. **Type soundness :** no well-typed program gets stuck on a non-value, and when a well-typed program terminates it gives a value of the right type;
2. **Termination :** every execution of a well-typed program ends with a value;
3. **Subject reduction :** the execution of a well-typed program preserves its type at each step;
4. **Determinism :** given an initial state, each well-typed program can only evaluate to one value and one new state;
5. **Type uniqueness :** each program has at most one type.

DEC supports Hoare logic reasoning, providing proven Hoare logic rules that will be detailed in section 3.3.3 p.27 ([16], THoare.v).

3.3.2 Deep embedding constructs

Before we move on to the main constructs of the deep embedding, we need to introduce some essential background types and structures ([16], StaticSemA.v) :

- **Id type :** represents the identifiers' type in the deep embedding. Typically, identifiers of variables are strings but we can use any other type;

- **Value type :** represents the actual values the programs will be manipulating. They can be built using the constructor *cst* followed by a shallow type and a value that should be of that type. The value type and its constructor are defined as follows :

Script 3.20: Values in the deep embedding

```
(** the internal type of values,
        parametrised by a semantic type *)
Inductive ValueI (T: Type) : Type := Cst (v: T).

(** the external type of values,
        hiding the semantic type *)
Definition Value : Type := sigT ValueI.

(** smart value constructor *)
Definition cst (T: Type) (v: T) : Value :=
          @existT Type ValueI T (Cst T v).
```

- **QValue type :** represents head normal forms for expressions (here called quasi-values) in the deep embedding which can be either actual values lifted to quasi-values using the constructor *QV* or identifiers of variables in the variable environment that we lift to quasi-functions using the constructor *Var*, as shown in the following script :

Script 3.21: Quasi-values in the deep embedding

```
Inductive QValue : Type := Var (x: Id)
                         | QV (v: Value).
```

- **Fun type :** represents functions in the deep embedding. The definition of this type is detailed in script 3.22 where :
  - ***fenv*** corresponds to the function environment which maps identifiers to functions;
  - ***tenv*** corresponds to the typing environment for parameters which maps identifiers to value types;
  - ***e0*** and ***e1*** are expressions in the deep embedding corresponding respectively to the base case and step case;
  - ***x*** corresponds to the name of the function;
  - ***n*** is the bound. If its value is 0, the first expression *e0* is evaluated else the second expression *e1* gets evaluated. This enables us to define terminating recursive functions.

Script 3.22: Functions in the deep embedding

```
Inductive Fun : Type :=
        FC (fenv: Envr Id Fun)
           (tenv: valTC)
           (e0 e1: Exp)
           (x: Id)
           (n: nat).
```

- **QFun type :** represents quasi-functions in the deep embedding which can be either actual functions that we lift to quasi-functions using the constructor *QF* or identifiers of functions in the function environment that we lift to quasi-functions using the constructor *FVar*, as shown in the following script :

Script 3.23: Quasi-functions in the deep embedding

```
Inductive QFun : Type := FVar (x: Id)
                       | QF (f: Fun).
```

- **XFun type :** represents generic effects (external stateful functions) in the deep embedding used to define prospective operations on the state and/or an input. It is used by the *Modify* construct of the deep embedding defined in script 3.28 p.26, when we want to read the state or perform changes on it and also when we need to implement external functions. As shown in script 3.24, it requires an input type *T1* as well as an output type *T2*. Thus, to define state operations, we will mainly need generic effects with *unit* as the input and output type. Moreover, when specifying a generic effect, we only need to worry about the b_mod attribute as b_exec and b_eval are already preset to build respectively the resulting state and value. This implementation deals with the issue of construct extensibility since it enables us to add shallow constructs and even use shallow functions :

Script 3.24: generic effects in the deep embedding

```
Record XFun (T1 T2: Type) : Type := {
    b_mod : W → T1 → prod W T2 ;
    b_exec : W → T1 → W :=
     fun state input ⇒ fst (b_mod state input) ;
    b_eval : W → T1 → T2 :=
     fun state input ⇒ snd (b_mod state input)
}.
```

As shown in script 3.28 ([16], StaticSemA.v), The deep embedding provides us with 8 constructs to build expressions :

1. **Val** *v* **:** lifts the value *v* to an expression using the constructor *Val*;
2. **BindN** *e1 e2* **:** represents a **sequence** of instructions where the expression *e1* is evaluated first then *e2* next. It is equivalent to "*e1;;e2*" in the shallow embedding;
3. **BindS** *x e1 e2* **:** represents an **assignment**. It is equivalent to "*perform x := e1 in e2*" in the shallow embedding. More precisely, the expression *e2* is evaluated in the updated variable environment where *x* gets mapped to the resulting value of the evaluation of *e1*;
4. **BindMS** *fenv env e* **:** the expression *e* is considered in the function environment *fenv* and variable environment *env*;
5. **IfThenElse** *e1 e2 e3* **:** represents a **conditional structure**. It is equivalent to "*if e1 then e2 else e3*" in the shallow embedding. This expression is considered well-typed only when *e1* gets evaluated to a deep boolean value;
6. **Return** *G qv* **:** lifts the quasi-value *qv* to an expression using the constructor *Return*. If *qv* is a variable, it gets evaluated in the variable environment. *G* is a flag which affirms whether the construct should get translated to an actual return in C or not. The type *Tag* of *G* is defined as follows :

Script 3.25: Tag type in the deep embedding

```
Inductive Tag : Type := LL | RR.
```

7. **Apply** *qf ps* **:** represents **function application**. If *qf* is a variable it gets evaluated in the function environment. *ps* is the list of parameters which are expressions. They should be well-typed relatively to the typing environment of the function i.e. the resulting value of the evaluation of each parameter's expression should match its awaited type preset in the typing environment. The type *Prms* of *ps* is defined as follows :

Script 3.26: Function parameters in the deep embedding

```
Inductive Prms : Type := PS (es: list Exp).
```

8. **Modify** *T1 T2 VT1 VT2 xf qv* **:** is used to define effectful state operations and functions. *xf* is the generic effect that specifies the operation, *T1* and *T2* are respectively its input and output types and *VT1* and *VT2* assure that *T1* and *T2* are admissible value types. If *qv* is a variable it gets evaluated in the variable environment. For instance, we can define a simple *SKIP* instruction with the *Modify* construct as follows :

Script 3.27: SKIP instruction in the deep embedding

```
Definition xf_skip : XFun unit unit := {|
        b_mod := fun state input => (state,tt) |}.

Definition SKIP : Exp := Modify unit unit
   UnitVT UnitVT xf_skip (QV(cst unit tt))
```

More examples of these constructs will be detailed in section 4 p.31.

Script 3.28: Deep embedding expressions

```
Inductive Exp : Type :=
 | Val (v: Value)
 | BindN (e1: Exp) (e2: Exp)
 | BindS (x: Id) (e1: Exp) (e2: Exp)
 | BindMS (fenv: Envr Id Fun) (env: valEnv) (e: Exp)
 | IfThenElse (e1: Exp) (e2: Exp) (e3: Exp)
 | Return (G: Tag) (qv: QValue)
 | Apply (qf: QFun) (ps: Prms)
 | Modify (T1 T2: Type) (VT1: ValTyp T1) (VT2: ValTyp T2)
              (xf: XFun T1 T2) (qv: QValue)
```

The deep expressions are meant to be evaluated in certain variable and function environments. To that end, single and multi-step evaluation were defined as follows ([16], DynamicSemA.v) :

- **EStep fenv env (Conf Exp n e) (Conf Exp n' e') :** represents a **single-step evaluation** of the expression *e*, in the function environment *fenv* and variable environment *env*, with *n* as the current state. The constructor *Conf* builds a configuration of the type we seek to evaluate which, in our case, is *Exp* and contains the current state as well as the starting expression. *n'* and *e'* are respectively the resulting state and expression.

- **PrmsStep fenv env (Conf Prms n es) (Conf Prms n' es') :** represents a single-step evaluation of the list of parameters *es* which conducts a single-step evaluation of the first expression in the list. If the first expression is a value, it moves on to the next parameter;
- **EClosure fenv env (Conf Exp n e) (Conf Exp n' e') :** represents a **multi-step evaluation** of the expression *e* which is defined as a reflexive transitive closure. It uses the predefined single evaluation steps. It is important to note that there exists a state where any expression gets evaluated to a certain value which we conclude from the termination property of the deep embedding. Furthermore, since the evaluation process is deterministic, an expression gets evaluated to a unique value.
- **PrmsClosure fenv env (Conf Prms n es) (Conf Exp n' es') :** represents a multi-step evaluation of the list of parameters *es* which conducts a multi-step evaluation of each expression in the list.

Finally, the state in the deep embedding is represented with a type parameter *W*, as shown in script 3.24 p.24, that can be specified as required.

3.3.3 Hoare logic in the deep embedding

A Hoare triple in the deep embedding is slightly different from its counterpart in the shallow embedding. Indeed, this new version of the Hoare triple depends on the function environment as well as the variable environment we are evaluating the expression in. Both will be passed to the Hoare triple as parameters along with the expression, the precondition and the postcondition. The precondition is a predicate on the state while the postcondition is a predicate on both the resulting value and state. We must also ensure that the expression we are dealing with is well-typed. Therefore, as shown in script 3.29 ([16], THoare.v), a Hoare triple is considered valid if and only if the expression *e* is well-typed and there is an *EClosure* from a certain starting state verifying the precondition to a certain resulting state and value that verify the postcondition with *e* evaluated in the given function and variable environments *fenv* and *env*. **{{P}} fenv >> env >> e {{Q}}** is the chosen notation for the Hoare triple in the deep embedding where *P*, *Q*, *e*, *fenv*, and *env* are respectively the precondition, the postcondition, the expression, the function environment and the variable environment. Furthermore, as the evaluation of parameters is different from that of expressions, we need to devise a separate Hoare triple for parameters as shown in script 3.30 where *EClosure* is replaced by *PrmsClosure*.

Script 3.29: Hoare triple for expressions in the deep embedding

```
Definition THoareTriple_Eval
    (P : W -> Prop) (Q : Value → W → Prop)
    (fenv: funEnv) (env: valEnv) (e: Exp) : Prop :=
∀ (ftenv: funTC) (tenv: valTC)
   (k1: FEnvTyping fenv ftenv)
   (k2: EnvTyping env tenv)
   (t: VTyp)
   (k3: ExpTyping ftenv tenv fenv e t)
   (s s': W) (v: Value),
 EClosure fenv env (Conf Exp s e) (Conf Exp s' (Val v))
 → P s → Q v s'.

Notation "{{ P }} fenv >> env >> e {{ Q }}" :=
(THoareTriple_Eval P Q fenv env e) (at level 90).
```

Script 3.30: Hoare triple for parameters in the deep embedding

```
Definition THoarePrmsTriple_Eval
    (P : W → Prop) (Q : list Value → W → Prop)
    (fenv: funEnv) (env: valEnv) (ps: Prms) : Prop :=
∀ (ftenv: funTC) (tenv: valTC)
   (k1: FEnvTyping fenv ftenv)
   (k2: EnvTyping env tenv)
   (pt: PTyp)
   (k3: PrmsTyping ftenv tenv fenv ps pt)
   (s s': W) (vs: list Value),
 PrmsClosure fenv env (Conf Prms s ps) (Conf Prms s'
        (PS (map Val vs))) → P s → Q vs s'.
```

Several Hoare logic rules were devised and proven ([16], THoare.v). We show the importance of these rules in section 4.6.1 p.52. For simplicity's sake, We will only give the formal definition of some of these rules and not their actual implementation in Coq :

- **weakening lemmas :** the weakening lemma on expressions is the same as the one defined for the shallow embedding in script 3.11 p.18. However, we need another weakening lemma for a Hoare triple on parameters. The weakest preconditions for expression and parameter triples were also defined as functions called respectively *wp* and *wpPrms*. The two weakening lemmas are formally defined as follows :

$$\frac{\{\{P'\}\}\ \text{fenv} >> \text{env} >> \text{e}\ \{\{Q\}\} \qquad P \rightarrow P'}{\{\{P\}\}\ \text{fenv} >> \text{env} >> \text{e}\ \{\{Q\}\}}\ \text{weakenEval}$$

$$\frac{\text{THoarePrmsTriple\_Eval P' Q fenv env ps} \qquad \text{P} \rightarrow \text{P'}}{\text{THoarePrmsTriple\_Eval P Q fenv env ps}}\ \text{weakenPrms}$$

- **BindS rule :** the assignment rule in the deep embedding is quite different from the shallow one, defined in script 3.12 p.18, as environments are now explicitly manipulated. Therefore, to prove a Hoare triple on an assignment of the form *BindS x e1 e2*, we need to prove a Hoare triple on *e2* where x is mapped to the resulting value of *e1* in the updated variable environment. To that end, we need an intermediate predicate that will ascertain the validity of the resulting value. This rule is formally defined as follows :

$$\frac{\begin{array}{c}\text{\{\{P0\}\} fenv >> env >> e1 \{\{P1\}\}} \\ \forall\ \text{v, \{\{P1 v\}\} fenv >> (x,v):env >> e2 \{\{P2\}\}}\end{array}}{\text{\{\{P0\}\} fenv >> env >> BindS x e1 e2 \{\{P2\}\}}}\ \text{BindS\_VHTT1}$$

- **BindN rule :** a variant of the *Hoare_seq* rule, introduced in section 3.2.1 p.15, which requires an intermediate predicate to prove a Hoare triple on a sequence of instructions. It is formally defined as follows :

$$\frac{\begin{array}{c}\text{\{\{P0\}\} fenv >> env >> e1 \{\{fun \_ } \Rightarrow \text{P1\}\}} \\ \text{\{\{P1\}\} fenv >> env >> e2 \{\{P2\}\}}\end{array}}{\text{\{\{P0\}\} fenv >> env >> BindN e1 e2 \{\{P2\}\}}}\ \text{BindN\_VHTT1}$$

- **IfThenElse rule :** a variant of the *Hoare_if* rule, introduced in section 3.2.1 p.15. In the deep embedding, we need an intermediate predicate to evaluate the condition then we reason on both possible cases. This rule is formally defined as follows :

$$\frac{\begin{array}{c}\text{\{\{P0\}\} fenv >> env >> e1 \{\{P1\}\}} \\ \text{\{\{P1 (cst bool true)\}\} fenv >> env >> e2 \{\{P2\}\}} \\ \text{\{\{P1 (cst bool false)\}\} fenv >> env >> e3 \{\{P2\}\}}\end{array}}{\text{\{\{P0\}\} fenv >> env >> IfThenElse e1 e2 e3 \{\{P2\}\}}}\ \text{IfTheElse\_VHTT1}$$

- **Apply rules :** the apply construct has several rules :
  - **Apply_VHTT1 :** the main rule of the Apply construct which evaluates not only the parameters but also the next recursive call of the function by performing a pattern matching on the bound. It is formally defined in Coq as follows :

Script 3.31: Main Hoare logic rule for the Apply construct

```
Lemma Apply_VHTT1 (P0: W → Prop)
   (P1: list Value → W → Prop)
   (P2: Value → W → Prop)
   (fenv: funEnv) (env: valEnv)
   (f: Fun) (es: list Exp) :
 THoarePrmsTriple_Eval P0 P1 fenv env (PS es) →
 match f with
  | FC fenv' tenv' e0 e1 x n ⇒
    length tenv' = length es /\
    match n with
    | 0 ⇒ (∀ vs: list Value,
       THoareTriple_Eval (P1 vs) P2 fenv'
           (mkVEnv tenv' vs) e0)
    | S n' ⇒ (∀ vs: list Value,
       THoareTriple_Eval (P1 vs) P2
           ((x,FC fenv' tenv' e0 e1 x n')::fenv')
           (mkVEnv tenv' vs) e1)
    end
  end →
 THoareTriple_Eval P0 P2 fenv env (Apply (QF f) (PS es)).
```

– **Apply_VHTT2 :** only evaluates the parameters in a function application. It is defined as follows :

$$\frac{\begin{array}{c}\text{THoarePrmsTriple\_Eval P0 P1 fenv env (PS es)}\\ \forall\ \text{vs, \{\{P1 vs\}\} fenv >> env >> Apply (QF f) (PS (map Val vs)) \{\{P2\}\}}\end{array}}{\text{\{\{P0\}\} fenv >> env >> Apply (QF f) (PS es) \{\{P2\}\}}}\ \text{Apply\_VHTT2}$$

– **QFun_VHTT :** evaluates a function variable in an *Apply* construct by searching its value in the function environment. It is formally defined as follows :

$$\frac{\begin{array}{c}\text{findET fenv x f}\\ \text{\{\{P\}\} fenv >> env >> Apply (QF f) (PS es) \{\{Q\}\}}\end{array}}{\text{\{\{P\}\} fenv >> env >> Apply (FVar x) (PS es) \{\{Q\}\}}}\ \text{QFun\_VHTT}$$

4. Proving invariants in the deep embedding

In this section, we show how we proved three different invariants of PIP. The companion code can be found at `https://github.com/CherifSami/coq_internship`. The first invariant is about a function that reads the memory. The second one is about a function that writes in the memory. The last invariant is about a recursive function. We explain throughout this section our approach in modelling these functions and the way we engineered our proofs while trying to make them modular and as simple as possible. The first section is dedicated to a briefing on the preliminary work we did to become more familiar with the deep embedding and the Hoare logic we built on progressively. In the second section, we explain how we modelled the PIP monad by replicating the PIP state. In later sections we show how we modelled the shallow functions and proved their invariants in the deep embedding. The last section is dedicated to our observations on the difference between shallow and deep proof as well as the limitations imposed on our verification proofs by the DEC language.

4.1 Preliminary experiments

For this preliminary work, we used an untyped form of the Hoare logic triple which was later refined to the one we defined in section 3.3.3 p.27. We started working with a natural number state on which we defined two functions. The first function called *ReadN* simply reads the current value of the state while the second one called *WriteN* writes a given value in the state. To model these functions, we used generic effects since they act directly on the state. We proved several Hoare triples about these functions. For instance, we proved that if we write a value x in the state then read it immediately after, we should get the same value x. This is quite similar to an invariant we proved on the PIP state, called *writeVirtualInvNewProp*, which is detailed in section 4.4 p.46. Then, we switched to an association list state which resembles more the PIP state. We also devised shallow reading and writing functions on this state as the ones defined on the PIP state in annex C p.63. We used these functions to define deep reading and writing functions on such state using generic effects. Then we proved similar lemmas to the one we did on the natural number state.

4.2 Modelling the PIP monad in the deep embedding

To prove invariants of PIP in the deep embedding, it is essential to replicate the PIP memory state specified in the PIP monad, defined in script 3.8 p.14. To that end, all type definitions mentioned in section 3.1.4 p.12 were copied in the file *PIP_state.v*. All the axioms, constructors, comparison functions as well as predefined values were also copied in this file as shown in annex B p.61. Then, we defined a module of type *IdModType*, detailed in annex A p.60, where the type parameter *W* is set to the PIP *state* record type, defined in script 3.1 p.12. We defined the initial value of this type parameter which corresponding to an empty memory. Furthermore, The *Id* type parameter for identifiers is set to *string*. This module, called *IdModP*, will be passed as a parameter to the modules we're going to work on later. It is defined in the the file *IdModPip.v* as follows :

Script 4.1: PIP state in the deep embedding

```
Require Import Pip_state.

Module IdModP <: IdModType.
  Definition Id := string.

  Definition W := state.

  Definition Loc_PI := valTyp_irrelevance.

  Definition BInit := {|
          currentPartition :=  defaultPage;
          memory:= @nil (paddr * value);
                      |}.
End IdModP.
```

In addition to the memory state, the PIP monad manages undefined behaviour. However, we don't have error handling in the deep embedding. Therefore, we are going to use *option* types to deal with undefined behaviour of the PIP monad. This will be further discussed in section 4.6.2 p.54.

4.3 1st invariant and proof

4.3.1 Invariant in the shallow embedding

This invariant concerns a function named *getFstShadow*. We want to prove that if the necessary properties for the correct execution of this function are verified then any precondition on the state persists after its execution since this function doesn't change it. We also need to ascertain the validity of the returned value. It is defined as follows ([13], Invariants.v) :

Script 4.2: getFstShadow invariant in the shallow embedding

```
Lemma getFstShadow (partition : page) (P : state → Prop) :
{{fun s ⇒ P s ⋀ partitionDescriptorEntry s ⋀
          partition ∈ (getPartitions multiplexer s) }}
Internal.getFstShadow partition
{{fun (sh1 : page) (s : state) ⇒ P s ⋀
          nextEntryIsPP partition sh1idx sh1 s }}.
```

where :

- **getFstShadow :** is a function that **returns the physical page of the first shadow** for a given partition. The index of the virtual address of the first shadow, called *sh1idx* as shown in annex B p.61, is predefined in PIP as the 4th index of a partition. Furthermore, we know that the virtual address and the physical address of any page are consecutive. Therefore, we only need to fetch the predefined index of the first shadow, calculate its successor then read the corresponding page in the given partition. It is defined as follows ([13], Internal.v) :

  Script 4.3: getFstShadow function in the shallow embedding

  ```
  Definition getFstShadow (partition : page):=
    perform idx := getSh1idx in
    perform idxsucc := MALInternal.Index.succ idx in
    readPhysical partition idxsucc.
  ```

- **P :** is the propagated property on the state;
- **getPartitions :** is a function that returns the list of all sub-partitions of a given partition. In our case, since we give it the multiplexer partition which is the root partition, it returns all the partitions in the memory. This function is used to verify that the partition we give to the *getFstShadow* function is valid by checking its presence in the partition tree;

- **nextEntryIsPP :** returns *True* if the entry at position successor of the given index in the given table is a physical page and is equal to another given page. It is defined as follows ([13], StateLib.v) :

Script 4.4: nextEntryIsPP property

```
Definition nextEntryIsPP table idxroot tableroot s : Prop:=
match Index.succ idxroot with
 | Some idxsucc =>
  match lookup table idxsucc (memory s) beqPage beqIndex with
   | Some (PP table) => tableroot = table
   |_ => False
  end
 | _ => False
end.
```

- **partitionDescriptorEntry :** defines some properties of the partition descriptor. All the predefined indexes in the file *PIPstate.v*, shown in annex B p.61, should be less than the table size minus one and contain virtual addresses. This is verified by the *isVA* property which returns *True* if the entry at the position of the given index in the given table is a virtual address and is equal to a given value. The successors of these indexes contain physical pages which should not be equal to the default page. This is verified by the *nextEntryIsPP* property. The *partitionDescriptorEntry* property is defined as follows ([13], Consistency.v) :

Script 4.5: partitionDescriptorEntry property

```
Definition partitionDescriptorEntry s :=
∀ (partition: page),
 partition ∈ (getPartitions multiplexer s) →
 ∀ (idxroot : index),
  (idxroot = PDidx ∨ idxroot = sh1idx ∨
   idxroot = sh2idx ∨ idxroot = sh3idx ∨
   idxroot = PPRidx ∨ idxroot = PRidx) →
  idxroot < tableSize - 1 ∧ isVA partition idxroot s ∧
  ∃ entry, nextEntryIsPP partition idxroot entry s ∧
           entry ≠ defaultPage.
```

In the next sections we will rewrite the *getFstShadow* function as well as adapt these properties in order to prove this invariant in the deep embedding.

4.3.2 Modelling the *getFstShadow* function

We worked on several possible definitions in the deep embedding for the *getFstShadow* function, defined in script 4.3 p.33 taking into account the limitations of the deep embedding especially in dealing with inductive data types as it doesn't provide us with a pattern matching construct.

getSh1idx function

First, let's define *getSh1idx* which returns the value of *sh1idx*. In the deep embedding, we defined it as a deep index value of the predefined first shadow index :

Script 4.6: getSh1idx definition in the deep embedding

```
Definition getSh1idx : Exp := Val (cst index sh1idx).
```

Index successor function

Next, we need to implement the index successor function in the deep embedding. An index value has to be less then the preset table size. This property should be verified before calculating the successor. This function is defined as follows in the shallow embedding ([13], MALInternal.v) :

Script 4.7: Index successor function in the shallow embedding

```
Program Definition succ (n : index) : LLI index :=
 let isucc := n+1 in
 if (lt_dec isucc tableSize)
 then ret (Build_index isucc _)
 else  undefined 28.
```

Although this function is defined in the hardware model, we chose to model it as part of our experimentation with the deep embedding. **Our approach is to rewrite this function in the deep embedding while leaving shallow bits that we progressively replace with deep definitions in order to make the shift from shallow proofs to deep ones gradual and modular**. First, we rewrote this function so that we could use it in a *Modify* construct replacing the output type by *option index*. We used the index constructor *CIndex* to build the new index. The new shallow version is defined as follows :

Script 4.8: Rewritten shallow index successor function

```
Definition succIndexInternal (idx:index) : option index :=
 let (i,_) := idx in
 if lt_dec i tableSize
 then Some (CIndex (i+1))
 else None.
```

Thus, the first version of the index successor function, called *Succ*, is defined as a *Modify* construct that uses a generic effect, called xf_succ, of input type *index* and output type *option index*, which calls the rewritten shallow function *succIndexInternal*. *Succ* is parametrised by the name of the input index variable which will be evaluated in the variable environment :

Script 4.9: Definition of Succ

```
Instance VT_index : ValTyp index.
Instance VT_option_index : ValTyp (option index).

Definition xf_succ : XFun index (option index) := {|
  b_mod := fun s (idx:index) ⇒ (s, succIndexInternal idx)
|}.

Definition Succ (x:Id) : Exp :=
  Modify index (option index) VT_index VT_option_index
                    xf_succ (Var x).
```

The second version of the successor function, called *SuccD*, is different from the former one. In this version we don't want to call the shallow function *succIndexInternal*. Instead, we are trying to devise a comparatively deeper definition of successor where the conditional structure is replaced by the deep construct *IfThenElse* and the assignments are defined using the *BindS* construct. The new version is defined as follows :

Script 4.10: Definition of SuccD

```
Definition SuccD (x:Id) :Exp :=
BindS "i" (prj1 x)
          (IfThenElse (LtDec "i" tableSize)
                      (Apply SomeCindexQF
                             (PS[SuccR "i"])
                      )
                      (Val(cst (option index) None))
          ).
```

where *prj1* is a projection of the first value of an index record which corresponds to the actual value of the index, *LtDec* is the definition of the comparison function using its shallow version, *SomeCindexQF* is a quasi-function that lifts a natural number to an *option index* typed value using the shallow constructors *Cindex* and *Some* successively and *SuccR* is a function that calculates the successor of a natural number. It is important to note that *SomeCindex* is defined as a *Modify* construct using a generic effect instead of a pure deep function since the deep embedding doesn't provide a pattern matching construct to deal with such cases. Their formal definitions in Coq are as follows :

Script 4.11: Functions called in SuccD

```
(* projection function *)
Definition xf_prj1 : XFun index nat := {|
   b_mod := fun s (idx:index) => (s,let (i,_) := idx in i)
|}.
Definition prj1 (x:Id) : Exp :=
  Modify index nat VT_index VT_nat xf_prj1 (Var x).

(* comparision function *)
Definition xf_LtDec (n: nat) : XFun nat bool := {|
   b_mod := fun s i ⇒ (s,if lt_dec i n then true else false)
|}.
Definition LtDec (x:Id) (n:nat): Exp :=
  Modify nat bool VT_nat VT_bool (xf_LtDec n) (Var x).

(* lifting funtion *)
Definition xf_SomeCindex : XFun nat (option index) := {|
   b_mod := fun s i ⇒ (s,Some (CIndex i))
|}.
Definition SomeCindex (x:Id) : Exp :=
  Modify nat (option index) VT_nat VT_option_index
              xf_SomeCindex (Var x).
Definition SomeCindexQF := QF
(FC emptyE [("i",Nat)] (SomeCindex "i")
    (Val (cst (option index) None)) "SomeCindex" 0).

(* successor function for natural numbers *)
Definition xf_SuccD : XFun nat nat := {|
   b_mod := fun s i => (s,S i)
|}.
Definition SuccR (x:Id) : Exp :=
  Modify nat nat VT_nat VT_nat xf_SuccD (Var x).
```

The last version calls a recursive function *plusR* that calculates the sum of two natural numbers. More precisely, we replace the call of *SuccR* in the former definition with *plusR 1* which adds one to its given parameter. *plusR* and the new successor function, called *SuccRec*, are defined as follows :

Script 4.12: Definition of PlusR

```
Definition plusR' (f: Id) (x:Id) : Exp :=
      Apply (FVar f) (PS [VLift (Var x)]).
Definition plusR (n:nat) := QF
(FC emptyE [("i",Nat)] (VLift(Var "i"))
    (BindS "p" (plusR' "plusR" "i") (SuccR "p")) "plusR" n).
```

Script 4.13: Definition of SuccRec

```
Definition SuccRec (x:Id) :Exp :=
BindS "i" (prj1 x)
          (IfThenElse (LtDec "i" tableSize)
                      (Apply SomeCindexQF
                             (PS[Apply (plusR 1)
                                       (PS[VLift(Var "i")])])
                      )
                      (Val(cst (option index) None))
          ).
```

readPhysical function

Now, we need to define the function that will read the physical page in the given index. This function, called *readPhysical* in the shallow embedding, uses the predefined lookup function that returns the value mapped to the address-index pair we give it. As shown in script 4.14 ([13], MAL.v), *readPhysical* checks whether the read page is actually a physical page by performing a match on the returned entry. *beqPage* and *beqIndex* are comparison functions respectively for pages and indexes.

Script 4.14: readPhysical function in the shallow embedding

```
Definition readPhysical (paddr: page) (idx: index) : LLI page:=
 perform s := get in
 let entry := lookup paddr idx s.(memory) beqPage beqIndex in
 match entry with
  | Some (PP a) ⇒ ret a
  | Some _ ⇒ undefined 5
  | None ⇒ undefined 4
 end.
```

To implement this function in the deep embedding, we first copied all the predefined association list functions in a file we named *Lib.v*, as shown in annex C p.63. We then rewrote this function so that we could use it in a *Modify* construct replacing the output type by *option page* as follows :

Script 4.15: Rewritten shallow readPhysical function

```
Definition readPhysicalInternal p i memory :option page :=
 match (lookup p i memory beqPage beqIndex) with
  | Some (PP a) ⇒ Some a
  | _ ⇒ None
 end.
```

The function is called *ReadPhysical* in the deep embedding and is parametrised by the name of the *option index* typed variable. *ReadPhysical* performs a match on its input to verify that it's a valid index then naturally calls *readPhysicalInternal*. It is defined as follows :

Script 4.16: Definition of ReadPhysical

```
Instance VT_option_page : ValTyp (option page).

Definition xf_read (p: page): XFun (option index) (option page) :=
{| b_mod := fun s oi => (s,match oi with
        |None => None
        |Some i => readPhysicalInternal p i (memory s) end) |}.

Definition ReadPhysical (p:page) (x:Id) : Exp :=
  Modify (option index) (option page)
         VT_option_index VT_option_page
         (xf_read p) (Var x).
```

Using these definitions we are going to define three versions of *getFstShadow* in the deep embedding, each calling a different version of the index successor function. These functions are named *getFstShadowBind*, *getFstShadowBindDeep* and *getFstShadowBindDeepRec* and respectively call Succ, SuccD and SuccRec. Since their definitions are same, we will only give the definition of *getFstShadowBind* :

Script 4.17: Definition of getFstShadowBind

```
Definition getFstShadowBind (p:page) : Exp :=
 BindS "x" getSh1idx
           (BindS "y" (Succ "x")
                      (ReadPhysical p "y")
           ).
```

4.3.3 Invariant in the deep embedding

To model this invariant in the deep embedding we will use the Hoare triple we defined in section 3.3.3 p.27. However, we need to adapt some of the properties mentioned in section 4.3.1 p.33. In particular, the property *nextEntryIsPP*, defined in script 4.4 p.34, needs to be parametrised by the resulting deep value. So the page we need to compare is of type *Value* instead of *page*. the comparison between the fetched and given value now becomes a comparison between two deep values and not shallow ones. Also, to calculate the successor of the given index we use the rewritten successor function *succIndexInternal*, defined in script 4.8 p.36. Also, when we call this property in *partitionDescriptorEntry*, defined in script 4.5 p.34, we need to lift the page to an *option page* typed deep value. This property is now defined as follows :

Script 4.18: Rewritten nextEntryIsPP property

```
Definition nextEntryIsPP (p:page) (idx:index) (p':Value) (s:W) :=
match succIndexInternal idx with
 | Some i =>
  match lookup p i (memory s) beqPage beqIndex with
   | Some (PP table) => p' = cst (option page) (Some table)
   |_ => False
  end
 | _ => False
end.
```

Finally, we can write the deep Hoare triple which is not only parametrised by the partition and the propagated property but also by the function environment as well as the variable environment we want to evaluate the *getFstShadow* function in. It is formally defined in Coq as follows :

Script 4.19: getFstShadow invariant definition

```
Lemma getFstShadowBindH (partition : page) (P : W -> Prop)
                        (fenv: funEnv) (env: valEnv) :
{{fun s ⇒ P s ∧ partitionDescriptorEntry s ∧
           partition ∈ (getPartitions multiplexer s)}}
fenv >> env >> (getFstShadowBind partition)
{{fun sh1 s ⇒ P s ∧ nextEntryIsPP partition sh1idx sh1 s}}.
```

Naturally, since we defined three *getFstShadow* functions, each calling a different version of the deep index successor function, we only need to specify the name of version of *getFstShadow* we want to call. In this case we are calling *getFstShadowBind* defined in script 4.17 p.39.

4.3.4 Invariant proof

Script 4.20: Proof of the getFstShadow invariant

```
Proof.
unfold getFstShadowBind. (* or other called function *)
eapply BindS_VHTT1.
eapply getSh1idxWp.
simpl; intros.
eapply BindS_VHTT1.
eapply weakenEval.
eapply succWp. (* or other Lemma for called function *)
simpl; intros; intuition.
instantiate (1:=(fun s => P s ∧ partitionDescriptorEntry s ∧
                  partition ∈ (getPartitions multiplexer s))).
simpl. intuition. instantiate (1:=sh1idx).
eapply H0 in H3.
specialize H3 with sh1idx.
eapply H3. auto. auto.
simpl; intros.
eapply weakenEval.
eapply readPhysicalW.
simpl;intros; intuition.
destruct H3. exists x.
unfold partitionDescriptorEntry in H1.
apply H1 with partition sh1idx in H4.
clear H1; intuition.
destruct H5. exists x0. intuition.
unfold nextEntryIsPP in H4.
unfold readPhysicalInternal; subst.
inversion H2.
repeat apply inj_pair2 in H3.
unfold nextEntryIsPP in H5.
rewrite H3 in H5.
destruct (lookup partition x (memory s) beqPage beqIndex).
unfold cst in H5.
destruct v0;try contradiction.
apply inj_pairT2 in H5.
inversion H5. auto.
unfold isVA in H4.
destruct (lookup partition sh1idx (memory s) beqPage beqIndex)
in H2; try contradiction. auto.
Qed.
```

As shown in the previous script, we start the proof by evaluating the first assignment using the assignment rule *BindS_VHTT1* defined in section 3.3.3 p.27. This implies evaluating *getSh1idx* and mapping its resulting value to x in the variable environment then evaluating the rest of the function in this updated environment. To evaluate *getSh1idx*, we use a lemma that we have proven called *getSh1idxWp*. This lemma also propagates any property on the state.

Script 4.21: getSh1idxWp lemma definition and proof

```
Lemma getSh1idxW (P: Value -> W -> Prop)
                 (fenv: funEnv) (env: valEnv) :
  {{wp P fenv env getSh1idx}} fenv >> env >> getSh1idx {{P}}.
Proof.
(* the weakest precondition is a precondition *)
apply wpIsPrecondition.
Qed.

Lemma getSh1idxWp P fenv env :
{{P}} fenv >> env >> getSh1idx
{{fun (idxSh1 : Value) (s : state) ⇒ P s
            ∧ idxSh1 = cst index sh1idx }}.
Proof.
eapply weakenEval. (* weakening precondition *)
eapply getSh1idxW.
intros.
unfold wp.
intros.
unfold getSh1idx in X.
inversion X;subst.
auto.
inversion X0.
Qed.
```

In script 4.3.1, two lines change in the the proof for the different *getFstShadow* functions. Indeed, in the first line, we need to adapt the name of the unfolded function according to the one we call in the Hoare triple definition. Later, we call the assignment rule again and we need to evaluate the successor function which is different in each *getFstShadow* definition. Therefore, we need to define a lemma for each version and call it when needed in the 8^{th} line. The version which corresponds to our case is defined and proven as follows :

Script 4.22: succWp lemma definition and proof

```
1  Lemma succWp (x:Id) (v:Value) P (fenv: funEnv) (env: valEnv) :
2   ∀ (idx:index),
3   {{fun s ⇒ P s ∧ idx < tableSize - 1 ∧ v=cst index idx}}
4    fenv >> (x,v)::env >> Succ x (* or other successor function *)
5   {{fun (idxsuc : Value) (s : state) ⇒ P s ∧
6           idxsuc = cst (option index) (succIndexInternal idx) ∧
7           ∃ i, idxsuc = cst (option index) (Some i)}}.
8  Proof.
9  intros.
10 eapply weakenEval.
11 eapply succW. (* or other lemma for called function *)
12 intros.
13 simpl.
14 split.
15 instantiate (1:=idx).
16 intuition.
17 intros.
18 intuition.
19 destruct idx.
20 exists (CIndex (i + 1)).
21 f_equal.
22 unfold succIndexInternal.
23 case_eq (lt_dec i tableSize).
24 intros.
25 auto.
26 intros.
27 contradiction.
28 Qed.
```

This lemma proves that we get a valid index after executing successor since the precondition assures its correct execution. Only the 11th line of the proof changes according the the called successor function. The lemma defined in the 11th line proves that the resulting value of the execution of the successor function is equal the the value we get when we apply the shallow *succIndexInternal* function to the same given index. The proof of this evaluation lemma is quite different between the various versions which is logical since evaluating a *Modify* that calls the shallow *succIndexInternal* function is different from evaluating all the deep constructs in the deep and recursive versions of the successor function.

The proof of the first lemma, called *succW*, which evaluates the *Succ* function defined in script 4.9 p.36, goes naturally by inversion on the closure. It is defined and proven as follows :

Script 4.23: succW Lemma definition and proof

```
Lemma succW  (x : Id) (P: Value → W → Prop) (v:Value)
             (fenv: funEnv) (env: valEnv) :
∀ (idx:index),
 {{fun s ⇒ idx < (tableSize -1) ∧ ∀ l : idx + 1 < tableSize,
    P (cst (option index) (succIndexInternal idx)) s ∧
    v = cst index idx }}
  fenv >> (x,v)::env >> Succ x {{ P }}.
Proof.
intros.
unfold THoareTriple_Eval; intros; intuition.
destruct H1 as [H1 H1'].
omega.
inversion X;subst.
inversion X0;subst.
repeat apply inj_pair2 in H7;subst.
inversion X2;subst.
inversion X3;subst.
inversion H;subst.
destruct IdModP.IdEqDec in H3.
inversion H3;subst.
clear H3 e X3 H XF1.
inversion X1;subst.
inversion X3;subst.
repeat apply inj_pair2 in H7.
repeat apply inj_pair2 in H9. subst.
unfold b_exec,b_eval,xf_succ,b_mod in *.
simpl in *.
inversion X4;subst.
apply H1.
inversion X5.
inversion X5.
contradiction.
Qed.
```

The proof of the second lemma, called *succDW*, which evaluates the *SuccD* function defined in script 4.10 p.36, is quite longer than the previous one. Indeed, we need to proceed by inversion on the evaluation of every deep construct. This lemma is defined and proven in annex D p.64.

For The third lemma, which evaluates the *SuccRec* function defined in script 4.13 p.38, we did two different proofs : one using only inversions, called *succRecWByInversion*, and the other using the Hoare logic rules shown in section 3.3.3 p.27, called *succRecW*. The former takes about 480 lines to prove while the latter takes approximatively 350 lines wich amounts to 130 lines less. Furthermore, the proof of the lemma *succRecW*, which uses Hoare logic rules seems more organised since we deal with the poof of each instruction at a time and not the function as a whole.

Finally, all what is left in the main proof is to evaluate the last function *ReadPhysical* and prove the implication between properties. To that end, we defined the following lemma, called *readPhysicalW* and proven in annex D p.64, as follows :

Script 4.24: readPhysicalW Lemma definition and proof

```
Lemma readPhysicalW (y:Id) table (v:Value)
        (P' : Value → W → Prop) (fenv: funEnv) (env: valEnv) :
 {{fun s ⇒ ∃ idxsucc p1, v = cst (option index) (Some idxsucc)
   ∧ readPhysicalInternal table idxsucc (memory s) = Some p1
   ∧ P' (cst (option page) (Some p1)) s}}
 fenv >> (y,v)::env >> ReadPhysical table y {{P'}}.
```

4.3.5 The Apply approach

For this approach, we chose to replace assignments in the *getFstShadow* function with function applications. For simplicity's sake, We chose to use the first version of the index successor function called *Succ* defined in script 4.9 p.36. However, considering our modular approach for its definition, we could easily replace it with its other versions and use the lemmas we devised for them in the new proof. To the *Apply* construct, we need to lift both *SuccD* and *readPhysical*, defined in script 4.16 p.39, to quasi-functions as follows :

Script 4.25: Lifting Succ and readPhysical to quasi-functions

```
Definition SuccQF := QF (FC emptyE [("y",indexType)]
 (Succ "y") (Val (cst (option index) None)) "Succ" 0).

Definition ReadPhysicalQF (p:page) := QF (FC
 emptyE [("x",optionIndexType)] (ReadPhysical p "x")
 (Val (cst (option page) None)) "ReadPhysical" 0).
```

The new version of the *getFstShadow* function, called *getFstShadowApply* is defined as follows :

Script 4.26: getFstShadowApply definition

```
Definition getFstShadowApply (p:page) :Exp :=
Apply (ReadPhysicalQF p) (PS [
                          Apply SuccQF (PS [getSh1idx])
                        ]).
```

For the proof, we didn't need to define any additional lemma about the intermediate functions. This is due to the use of the Hoare logic rules. The new invariant, called *getFstShadowApplyH'*, is defined and proven in annex D p.64.

4.4 2nd invariant and proof

This new invariant is about a function called *writeVirtual* that writes a virtual address in the memory. We want to prove that this function verifies all of PIP's properties which include memory isolation, vertical sharing, kernel data isolation and consistency properties, mentioned in section 3.1.2 p.6. In the shallow embedding, *writeVirtual* is defined as follows ([13], MAL.v) :

Script 4.27: writeVirtual function in the shallow embedding

```
Definition writeVirtual (paddr: page) (idx: index)
                        (va: vaddr) :LLI unit :=
 modify (fun s ⇒ {|
  currentPartition := s.(currentPartition);
  memory := add paddr idx (VA va) s.(memory) beqPage beqIndex
 |}).
```

First, we rewrote this function to act directly on the state as follows :

Script 4.28: Rewritten shallow writeVirtual function

```
Definition writeVirtualInternal (p:page) (i:index) (v:vaddr) :=
 fun s ⇒ {|
  currentPartition := s.(currentPartition);
  memory := add p i (VA v) s.(memory) beqPage beqIndex |}.
```

It is clear that the *writeVirtual* function is purely a generic effect. Its output value is irrelevant so its output type is declared as *unit*. We only need to call the *writeVirtualInternal* function in the generic effect which will be used the *Modify* construct to define the deep version, called *WriteVirtual*, as follows :

Script 4.29: WriteVirtual definition

```
Definition xf_writeVirtual (p: page) (i: index) (v: vaddr)
                           : XFun unit unit :=
 {| b_mod := fun s _ ⇒ (writeVirtualInternal p i v s,tt) |}.

Definition WriteVirtual (p: page) (i: index) (v: vaddr) : Exp :=
  Modify unit unit VT_unit VT_unit (xf_writeVirtual p i v)
                   (QV (cst unit tt)).
```

To check that our implementation is correct, we first focused on a simpler invariant which asserts a relevant property of the *writeVirtual* function. This property is included in the postcondition of the main invariant we want to prove. Indeed, when we write a virtual address v in the page p in the memory, at a certain position, and when we read the page p immediately after, at the exact same position, we should get the same value v. This Lemma is defined and proven as follows :

Script 4.30: writeVirtualInvNewProp invariant definition

```
Lemma writeVirtualInvNewProp (p : page) (i:index) (v:vaddr)
                             (fenv: funEnv) (env: valEnv) :
 {{fun _ ⇒ True}}
  fenv >> env >> WriteVirtual p i v
 {{fun _ s ⇒ readVirtualInternal p i s.(memory) = Some v}}.
Proof.
unfold THoareTriple_Eval; intros.
clear H k3 t k2 k1 tenv ftenv.
inversion X;subst.
inversion X0;subst.
repeat apply inj_pair2 in H5.
apply inj_pair2 in H7.
subst.
unfold b_eval, b_exec, xf_writeVirtual, b_mod in *.
simpl in *.
inversion X1;subst.
unfold writeVirtualInternal;simpl.
unfold add.
unfold readVirtualInternal;simpl.
specialize beqPairsTrue with p i p i.
intros;intuition.
rewrite H. reflexivity.
inversion X2.
inversion X2.
Qed.
```

To define the main invariant, we copied all the definitions of PIP's propagated properties as well as PIP's internal and dependant-type lemmas respectively in the files *Pip_Prop.v Pip_InternalLemmas.v* and *Pip_DependentTypeLemmas.v*. Then we defined and proved the following lemma about the *writeVirtual* function :

Script 4.31: writeVirtualWp lemma definition and proof

```
Lemma writeVirtualWp (p: page) (idx: index) (vad: vaddr)
      (P: Value → state → Prop) (fenv: funEnv) (env: valEnv) :
 {{fun  s ⇒ P (cst unit tt) {|
   currentPartition := currentPartition s;
   memory := add p idx (VA vad) (memory s) beqPage beqIndex |} }}
 fenv >> env >> WriteVirtual table idx addr  {{P}}.
Proof.
unfold THoareTriple_Eval.
intros.
inversion X;subst.
inversion X0;subst.
repeat apply inj_pair2 in H6.
repeat apply inj_pair2 in H8.
subst.
unfold xf_writeVirtual, b_eval, b_exec, b_mod in *.
simpl in *.
inversion X1;subst.
auto.
inversion X2.
inversion X2.
Qed.
```

Finally, we use the lemma above to prove the main invariant. As expected, the deep proof is practically identical to the shallow proof since we're weakening the Hoare triple first to use the *writeVirtualWp* lemma then we're proving a direct implication between properties on the state. The main invariant, called *writeVirtualInv*, is defined and proven in annex E p.71.

4.5 3rd invariant and proof

The third invariant is about a recursive function of PIP called *initVAddrTable* that initializes virtual addresses of a given table to the default value *defaultVAddr* defined in annex B p.61. This function is defined as follows in the shallow embedding ([13], Internal.v) :

Script 4.32: initVAddrTable in the shallow embedding

```
Fixpoint initVAddrTableAux timeout shadow2 idx :=
  match timeout with
  | 0 ⇒ ret tt
  | S timeout1 ⇒
    perform maxindex := getMaxIndex in
    perform res := MALInternal.Index.ltb idx maxindex in
    if (res)
    then
      perform defaultVAddr := getDefaultVAddr in
      writeVirtual shadow2 idx defaultVAddr;;
      perform nextIdx :=  MALInternal.Index.succ idx in
      initVAddrTableAux timeout1 shadow2 nextIdx
    else
      perform defaultVAddr := getDefaultVAddr in
      writeVirtual shadow2 idx defaultVAddr
  end.

(* Specifies the timeout of  initVAddrTableAux *)
Definition initVAddrTable sh2 n :=
  initVAddrTableAux tableSize sh2 n.
```

where :

- **getMaxIndex :** returns the value of the maximum index which is equal to the table size minus one, knowing that the table size is different than 0. We wrote the following simplified definition to use in the deep embedding :

  Script 4.33: maxIndex definition

  ```
  Axiom tableSizeNotZero : tableSize <> 0.
  Definition maxIndex : index := CIndex(tableSize-1).
  ```

- **succ :** is the index successor function defined in script 4.7 p.35. We will replace it with its deep implementation *SuccD* defined in script 4.10 p.36;

- **ltb :** is a comparison function for indexes similar to the mathematical comparison operator $<$ for natural numbers. In the deep embedding, we will define this function by calling its shallow version in as a generic effect as follows :

Script 4.34: LtLtb definition

```
Definition xf_Ltb (i:index) : XFun index bool :=
 {| b_mod := fun s idx => (s,Index.ltb idx i) |}.

Definition LtLtb (x:Id) (i:index) : Exp :=
  Modify index bool VT_index VT_bool (xf_Ltb i) (Var x).
```

- **writeVirtual :** is the function, defined in script 4.27 p.46, that writes a virtual address in the memory. We couldn't use its previous implementation, called *WriteVirtual*, defined in script 4.29 p47 because we're not working directly with the value of the given index but with a variable which we need to evaluate in the variable environment. The new version of this function called *writeVirtual'* is defined as follows :

Script 4.35: writeVirtual new definition

```
Definition xf_writeVirtual' (p:page) (v:vaddr) :XFun index unit:=
  {| b_mod := fun s i => (writeVirtualInternal p i v s,tt) |}.

Definition WriteVirtual' (p:page) (i:Id) (v:vaddr) :Exp :=
 Modify index unit VT_index VT_unit (xf_writeVirtual' p v) (Var i).
```

We also need to define *ExtractIndex* which extracts an index from an *option index* typed value by performing a pattern matching. This is only possible by using a generic effect as follows :

Script 4.36: ExtractIndex definition

```
Definition xf_ExtractIndex : XFun (option index) index :=
{| b_mod := fun s idx => (s,match idx with
                           |Some i => i
                           |_ => index_d end) |}.

Definition ExtractIndex (x:Id):=
  Modify (option index) index VT_option_index VT_index
              xf_ExtractIndex (Var x).
```

To simplify the proof and avoid repetition, we will place the instruction that calls *WriteVirtual'* before the conditional structure. *initVAddrTableAux* and *initVAddrTable* are defined as follows :

Script 4.37: initVAddrTable definition in the deep embedding

```
Definition initVAddrTableAux (f i: Id) (p:page) : Exp :=
BindN (WriteVirtual' p i defaultVAddr)
      (IfThenElse (LtLtb i maxIndex)
                  (BindS "y" (BindS "idx" (SuccD i)
                                          (ExtractIndex "idx"))
                             (Apply (FVar f)
                                    (PS [VLift(Var "y")])))
                  (Val (cst unit tt))).
Definition initVAddrTable (p:page) (i:index) :=
Apply (QF (FC emptyE [("x",Index)] (Val (cst unit tt))
          (initVAddrTableAux "initVAddrTable" "x" p)
          "initVAddrTable" tableSize) (PS[Val (cst index i)]).
```

The invariant we want to prove, called *initVAddrTableNewProperty*, is defined in script 4.38. The proof is done by induction on the bound. More precisely, we suppose that the Hoare triple is valid for the bound n and we need to prove it's valid for $S\ n$. We mainly used the Hoare logic rules defined in section 3.3.3 p.27 to evaluate each deep construct to get to the next function application where n is the bound which we have as a hypothesis. The main difficulty was to get to the exact same expression as well as the same function and variable environments as we have in the hypothesis without unfolding the Hoare triple. Indeed, if we unfold the Hoare triple at some point earlier, we have to reason by inversion on large expressions and we may encounter some typing problems which makes the proof much more complicated. In order to break down the proof of the step case, we defined a new lemma about the successor function, called *succWp'*, similar to the *succWp* lemma defined in script 4.22 p.43, in which we propagate the specification of the value added to the environment. It is also important to note that, to prove *succWp'*, we used *succDW*, the evaluation lemma for the index successor function defined and proven in annex D p.64, which reinforces the importance of our modular approach. *succWp'* and the proof of the *initVAddrTableNewProperty* lemma is detailed in annex F p.74.

Script 4.38: initVAddrTableNewProperty invariant in the deep embedding

```
Lemma initVAddrTableNewProperty table (curidx : index)
                  (fenv: funEnv) (env: valEnv) :
 {{ fun s ⇒ (∀ idx : index, idx < curidx →
     (readVirtual table idx (memory s) = Some defaultVAddr) )}}
  fenv >> env >> initVAddrTable table curidx
 {{fun _ s ⇒ ∀ idx, readVirtual table idx s.(memory) =
                      Some defaultVAddr }}.
```

4.6 Observations

4.6.1 Deep vs shallow

Specifying programs in the deep embedding has the great advantage of simplifying their syntactic manipulation. It also ensures a stricter structuring of program expressions. An interesting conjecture is whether this structure is reflected in the proofs. If we consider naive proofs mostly done by applying inversions, this is not the case. Such proofs can be really monotonous but overall they lack structure and readability since we deal with the program as a whole and not each instruction at a time like we do in the shallow proofs. However using the Hoare rules, shown in section 3.3.3 p.27, we could make proofs that compare favourably with the shallow ones in terms of structuring. Nevertheless, they become comparatively longer. Furthermore, proof specifications are generally more complex in the deep embedding as we need to worry about the specification of deep values among others.

Proof structuring

Importance of Hoare logic rules Throughout our experiments, the Hoare logic rules shown in section 3.3.3 p.27 have proven efficient at dealing with proofs in the deep embedding and this is due to many reasons :

- They enable us to structurally decompose our proofs and deal with each instruction at a time which makes them well structured and legible. This is really important when dealing with large expressions and it is what made our compositional approach possible.
- They enable us to construct shorter proofs as was the case with the *succRecW* lemma, mentioned in section 4.3.4 p.41, which was 130 lines less than its counterpart with inversions only.
- They internally deal with well-typedness issues which simplifies further our proofs.

To summarise, Hoare logic rules made our proofs **simpler**, **shorter**, **more structured** and **legible**. That's why we strongly advise their use in deep proofs.

Proof specifications

Dealing with values in the deep embedding Values in the deep embedding encapsulate their shallow type and value. This has a big impact on deep proofs. Indeed, when we need to communicate a value to certain lemma, we would naturally parametrise the lemma by a *Value* typed parameter as in lemma *succWp*, defined in script 4.22 p.43. In script 4.39, we give a slightly different definition of this lemma in which we omit the value specification part in the precondition. Proving this lemma is impossible as we don't have any information about the deep value *v*. Thus, we can't ascertain that it is *index* typed as a deep value nor that its actual value is *idx*. This differs from shallow proofs where we manipulate directly shallow typed values. Furthermore, this makes dealing with variable and function environments more complicated as we may have to propagate some values in the environments as we did with lemma *succWp'* mentioned in section 4.5 p.48.

Script 4.39: succWp false lemma definition

```
1 Lemma succWp (x:Id) (v:Value) P (fenv: funEnv) (env: valEnv) :
2  ∀ (idx:index),
3  {{fun s ⇒ P s ∧ idx < tableSize - 1}}
4   fenv >> (x,v)::env >> Succ x (* or other successor function *)
5  {{fun (idxsuc : Value) (s : state) ⇒ P s ∧
6          idxsuc = cst (option index) (succIndexInternal idx) ∧
7          ∃ i, idxsuc = cst (option index) (Some i)}}.
```

Defining Shallow functions As there is a dimorphism between the deep embedding and the logic, it is generally necessary to have both a shallow and deep version of each function. The shallow versions are mainly used in the predicates included in the preconditions and postconditions of Hoare triples. For example, we used the shallow *succIndexInternal* function, defined in script 4.8 p.36, in the postcondition of the *succWp* lemma, defined in script 4.22 43, to describe the resulting value of the deep index successor function. We also used it extensively in the proof of the third invariant *initVAddrTableNewProperty*, detailed in annex F p.74. This practically implies double amount of work for deep proofs compared to shallow ones where we use only shallow functions. A possible solution to this problem is to define a deep language with embedded logic [10, Hoare Logic (Part II)].

4.6.2 Limitations imposed by DEC

DEC was intended to support translation and syntactic manipulation rather than modelling and verification. Therefore, some limitations were imposed by DEC on our verification proofs. Indeed, DEC doesn't provide explicit error handling. For this reason we used *option* types. However, this use involved pattern matching. The deep embedding doesn't provide us with a pattern matching construct which makes dealing with inductive types rather intricate. Indeed, we have to deal with such types at the shallow level, which is possible using generic effects. This is clearly outlined in the definition of *ReadPhysical* in script 4.14 p.38, where we had to specify its input type as *option index* instead of *index* and perform the pattern matching before reading the state. In this case, we couldn't put the pattern matching in a separate generic effect since we would have had an issue with the *None* case. Another problem which arises in this case is that we may need to duplicate functions if, for instance, we need to use a definition of *ReadPhysical* without pattern matching. **An effective but challenging solution to this problem would be to devise a pattern matching construct in the deep embedding**. Nevertheless, it is important to note that, if we still intend for DEC to support only translation and syntactic manipulation, the lack of a pattern matching construct will not be an issue since there aren't inductive types in C.

4.6.3 Further discussion about the deep implementation of shallow functions

What should we consider as a sufficiently deep implementation of a shallow function ? The deep embedding of PIP is implemented in a way that allows extensibility of deep constructs using generic effects. This enabled us to model our functions gradually and temporarily resolved the problem of the lack of a pattern matching construct. But, it still remains quite risky since we can confuse deep with shallow. Indeed, the first version of the index successor function, called *Succ* and defined in script 4.9 p.36, is merely a call to the shallow function *succIndexInternal*, defined in script 4.8 p.36, that we wrapped in the *Modify* construct. Although, *Succ* was defined as a deep expression, it is just the outer shell of the function that is deep but structurally it is shallow.

Then, we devised a comparatively deeper version of this function, called *SuccD* and defined in script 4.10 p.36. But, there are still shallow bits in this version like the index comparison function in particular. **In a completely deep definition of a shallow function, generic effects should be used just for essentially stateful operations.** However, the degree to which we should deepen a shallow function is a choice that must be set according to our needs. For example, in our experiments we chose to omit completely deepening comparison functions. In the case of the PIP protokernel, deepening a shallow function depends on whether it is part of the algorithmic PIP code or the hardware model shown in figure 3.4 p.7.

5. Conclusion

In this report, we explained how we proved three different invariants of PIP in the deep embedding after modelling their corresponding functions while following a modular approach. The results we obtained were interesting with respect to our initial conjecture concerning proof structuring. Indeed, although the deep embedding ensures a stricter structuring of program expressions, this structure isn't reflected in naive proofs done by inversion. However, by using Hoare logic rules we managed to make our proofs more structured than shallow ones by decomposing our programs to simple instructions instead of considering them as a whole. Nevertheless, specifications of such proofs remain more complicated than the ones done in the shallow embedding. *Wildmoser* and *Nipkow* identified the same disadvantage while working on a deep assembly language using the Isabelle proof assistant [17]. Aside from the main differences between deep and shallow proofs, we also pointed out the limitations imposed on our verification proofs by DEC.

Many different adaptations, tests, and experiments have been left for the future due to lack of time. Possible future work may also involve the automation of proofs in the deep embedding as they seemed monotonous when done by inversion and predictable when it came to applying Hoare logic rules as each one was devised for a specific instruction of DEC. This is also possible since we are dealing with a closed language.

Logic theory and formal proofs in particular remain two of the most currently researched scientific fields. Indeed, proving that programs are correct became essential especially when we are dealing with critical systems. It is also important to make these proofs as structured as possible in order to make them legible and amendable. This is also crucial if we want to simplify further these proofs.

This internship was a beneficial first-hand research experience in such a renowned laboratory as CRIStAL. And although many technical difficulties were encountered, overcoming those challenges was the best way to learn. This experience was not only constructive but also enjoyable as the atmosphere within the team was both professional and friendly.

Bibliography

[1] Churchill, M., Mosses, P. D., Sculthorpe, N., and Torrini, P. "Reusable Components of Semantic Specifications". In: *TAOSD*. LNCS 8989 12. Springer, 2015, pp. 132–179.

[2] *CRIStAL*. Official website. URL: `https://www.cristal.univ-lille.fr/`.

[3] Dijkstra, E. W. *A discipline of programming*. Prentice-Hall Englewood Cliffs, N.J, 1976. ISBN: 013215871.

[4] Hoare, C. A. R. "An Axiomatic Basis for Computer Programming". In: *Commun. ACM* 12.10 (Oct. 1969), pp. 576–580. ISSN: 0001-0782. DOI: `10.1145/363235.363259`. URL: `http://doi.acm.org/10.1145/363235.363259`.

[5] Jomaa, N., Nowak, D., Grimaud, G., and Hym, S. "Formal Proof of Dynamic Memory Isolation Based on MMU". In: *Science of Computer Programming* (2017). ISSN: 01676423.

[6] Jomaa, N., Grimaud, G., Hym, S., and Nowak, D. "Coding a real OS kernel in Coq and formalising its security (unpublished paper draft)". Jan. 2017.

[7] Jomaa, N., Grimaud, G., Hym, S., and Nowak, D. "PIP : A Minimal Kernel with Provable isolation (unpublished paper draft)". 2017.

[8] Nowak, D. *Pip: A Minimal OS Kernel with Provable Isolation*. Third French-Japanese Meeting on Cybersecurity. Apr. 2017. URL: `https://project.inria.fr/FranceJapanICST/files/2017/05/DNowak_presentation_2017.pdf`.

[9] Paulin-Mohring, C. "Introduction to the Calculus of Inductive Constructions". In: *All about Proofs, Proofs for All*. Ed. by B. W. Paleo and D. Delahaye. Vol. 55. Studies in Logic (Mathematical logic and foundations). College Publications, 2014. Chap. Structural abstract interpretation, A formal study using Coq. ISBN: 978-1-84890-166-7.

[10] Pierce, B. C. et al. *Software Foundations*. Ebook version 4.2. Electronic textbook, 2017. URL: http://www.cis.upenn.edu/~bcpierce/sf.

[11] Plotkin, G. D. "A Structural Approach to Operational Semantics". In: *J. Log. Algebr. Program.* 60-61 (2004), pp. 17–139.

[12] Tanenbaum, A. "Modern Operating Systems". In: 3rd. Pearson, 2009. Chap. History of Operation Systems, pp. 60–69. ISBN: 013359162X.

[13] The 2XS team. *The PIP protokernel official Website*. CRIStAL laboratry. URL: http://pip.univ-lille1.fr.

[14] The Coq Development Team. *The Coq Proof Assistant. Reference Manual*. Coq version 8.6.1. INRIA, July 2017. URL: https://coq.inria.fr/distrib/current/files/Reference-Manual.pdf.

[15] The Coq development team. *Coq official website*. Inria. URL: https://coq.inria.fr.

[16] Torrini, P. and Nowak, D. "The DEC language repository". Aug. 2017. URL: https://github.com/ptorrx/DeepC4Pip.

[17] Wildmoser, M. and Nipkow, T. "Certifying Machine Code Safety: Shallow versus Deep Embedding". In: *Theorem Proving in Higher Order Logics (TPHOLs 2004)*. Ed. by K. Slind, A. Bunker, and G. Gopalakrishnan. Vol. 3223. 2004, pp. 305–320.

Appendixes : Project Files

A. IdModTypeA.v

```
1  (* Imports ... *)
2
3  Module Type IdModType.
4
5  Parameter Id : Type.
6
7  Parameter IdEqDec : forall (x y : Id), {x = y} + {x <> y}.
8
9  Instance IdEq : DEq Id :=
10 {
11   dEq := IdEqDec
12 }.
13
14 Parameter W : Type.
15
16 Parameter Loc_PI : forall (T: Type) (p1 p2: ValTyp T), p1 = p2.
17
18 Parameter BInit : W.
19
20 Instance WP : PState W :=
21 {
22   loc_pi := Loc_PI;
23
24   b_init := BInit
25 }.
26
27 End IdModType.
```

B. Pip_state.v

```
(* Imports ... *)

(* PIP axioms *)
Axiom tableSize nbLevel nbPage: nat.
Axiom nbLevelNotZero: nbLevel > 0.
Axiom nbPageNotZero: nbPage > 0.
Axiom tableSizeIsEven : Nat.Even tableSize.
Definition tableSizeLowerBound := 14.
Axiom tableSizeBigEnough : tableSize > tableSizeLowerBound.

(* Type definitions ... *)

(*Constructors*)
Parameter index_d : index.
Parameter page_d : page.
Parameter level_d : level.

Program Definition CIndex  (p : nat) : index :=
 if (lt_dec p tableSize)
 then Build_index p _
 else index_d.

Program Definition CPage (p : nat) : page :=
 if (lt_dec p nbPage)
 then Build_page p _
 else page_d.

Program Definition CVaddr (l: list index) : vaddr :=
 if ( Nat.eq_dec (length l)  (nbLevel+1))
 then Build_vaddr l _
 else Build_vaddr (repeat (CIndex 0) (nbLevel+1)) _.

Program Definition CLevel ( a :nat) : level :=
 if lt_dec a nbLevel
 then  Build_level a _
 else level_d.

(* Comparison functions *)
Definition beqIndex (a b : index) : bool := a =? b.
Definition beqPage (a b : page) : bool := a =? b.
```

```
41 Definition beqVAddr (a b : vaddr) : bool := eqList a b beqIndex.
42
43 (* Predefined values *)
44 Definition multiplexer := CPage 1.
45 Definition PRidx := CIndex 0.   (* descriptor *)
46 Definition PDidx := CIndex 2.   (* page directory *)
47 Definition sh1idx := CIndex 4.  (* shadow1 *)
48 Definition sh2idx := CIndex 6.  (* shadow2 *)
49 Definition sh3idx := CIndex 8.  (* configuration pages *)
50 Definition PPRidx := CIndex 10. (* parent *)
51
52 Definition defaultIndex := CIndex 0.
53 Definition defaultVAddr := CVaddr (repeat (CIndex 0) nbLevel).
54 Definition defaultPage := CPage 0.
55 Definition fstLevel :=  CLevel 0.
56 Definition Kidx := CIndex 1.
```

C. Lib.v

```
(* Imports ... *)

Fixpoint eqList {A : Type} (l1 l2 : list A)
         (eq : A -> A -> bool) : bool :=
 match l1, l2 with
  |nil,nil => true
  |a::l1' , b::l2' => if  eq a b then eqList l1' l2' eq else false
  |_ , _ => false
 end.

Definition beqPairs {A B: Type} (a : (A*B)) (b : (A*B))
           (eqA : A -> A -> bool) (eqB : B -> B -> bool) :=
 if (eqA (fst a) (fst b)) && (eqB (snd a) (snd b))
 then true else false.

Fixpoint lookup {A B C: Type} (k : A) (i : B) (assoc : list
  ((A * B)*C)) (eqA : A -> A -> bool) (eqB : B -> B -> bool) :=
 match assoc with
  | nil => None
  | (a, b) :: assoc' => if beqPairs a (k,i) eqA eqB
         then Some b else lookup k i assoc' eqA eqB
 end.

Fixpoint removeDup {A B C: Type} (k : A) (i : B) (assoc : list
  ((A * B)*C) )(eqA : A -> A -> bool) (eqB : B -> B -> bool)   :=
 match assoc with
  | nil => nil
  | (a, b) :: assoc' => if beqPairs a (k,i) eqA eqB
        then removeDup k i assoc' eqA eqB
        else (a, b) :: (removeDup k i assoc' eqA eqB)
 end.

Definition add {A B C: Type} (k : A) (i : B) (v : C) (assoc : list
  ((A * B)*C) ) (eqA : A -> A -> bool) (eqB : B -> B -> bool)  :=
 (k,i,v) :: removeDup k i assoc eqA eqB.

Definition disjoint {A : Type} (l1 l2 : list A) : Prop :=
forall x : A,  In x l1  -> ~ In x l2 .
```

D. Hoare_getFstShadow.v

```
(* Imports ... *)

(* Definitions & lemmas ...*)

Lemma succDW  (x : Id) (P: Value → W → Prop) (v:Value)
              (fenv: funEnv) (env: valEnv) :
∀ (idx:index),
 {{fun s ⇒ idx < (tableSize -1) ∧ ∀ l : idx + 1 < tableSize,
    P (cst (option index) (succIndexInternal idx)) s ∧
    v = cst index idx }}
  fenv >> (x,v)::env >> SuccD x {{ P }}.
Proof.
intros.
unfold THoareTriple_Eval; intros.
clear k3 t k2 k1 tenv ftenv.
intuition.
destruct H1 as [H1 H1'].
omega.
inversion X;subst.
inversion X0;subst.
inversion X2;subst.
repeat apply inj_pair2 in H7;subst.
inversion X3;subst.
inversion X4;subst.
inversion H;subst.
destruct IdModP.IdEqDec in H3.
inversion H3;subst.
clear H3 e X4 H XF1.
inversion X1;subst.
inversion X4;subst.
inversion X6;subst.
repeat apply inj_pair2 in H7.
repeat apply inj_pair2 in H9.
subst.
unfold xf_prj1 at 3 in X6.
unfold b_exec,b_eval,b_mod in *.
simpl in *.
destruct idx.
inversion X5;subst.
inversion X7;subst.
```

```
41 inversion X8;subst.
42 inversion X9;subst.
43 simpl in *.
44 inversion X11;subst.
45 inversion X12;subst.
46 repeat apply inj_pair2 in H7.
47 subst.
48 inversion X13;subst.
49 inversion X14;subst.
50 inversion H;subst.
51 clear H X14 XF2.
52 inversion X10;subst.
53 inversion X14;subst.
54 simpl in *.
55 inversion X16;subst.
56 inversion X17;subst.
57 repeat apply inj_pair2 in H7.
58 repeat apply inj_pair2 in H10.
59 subst.
60 unfold xf_LtDec at 3 in X17.
61 unfold b_exec,b_eval,b_mod in *.
62 simpl in *.
63 case_eq (lt_dec i tableSize).
64 intros.
65 rewrite H in X17, H1.
66 inversion X15;subst.
67 inversion X18;subst.
68 simpl in *.
69 inversion X20; subst.
70 inversion X19;subst.
71 inversion X21;subst.
72 simpl in *.
73 inversion X23;subst.
74 inversion H10;subst.
75 destruct vs; inversion H2.
76 inversion X24;subst.
77 inversion X25;subst.
78 repeat apply inj_pair2 in H11.
79 subst.
80 inversion X26;subst.
81 inversion X27;subst.
82 inversion H2;subst.
83 clear X27 H2 XF3.
```

```
84  inversion X22;subst.
85  inversion X27;subst.
86  simpl in *.
87  inversion X29;subst.
88  inversion H10;subst.
89  destruct vs; inversion H2.
90  inversion X30;subst.
91  inversion X31;subst.
92  repeat apply inj_pair2 in H11.
93  repeat apply inj_pair2 in H13.
94  subst.
95  unfold xf_SuccD at 3 in X31.
96  unfold b_exec,b_eval,xf_SuccD,b_mod in *.
97  simpl in *.
98  inversion X28;subst.
99  inversion X32;subst.
100 simpl in *.
101 inversion X34;subst.
102 inversion H10;subst.
103 destruct vs.
104 inversion H2.
105 inversion H2;subst.
106 destruct vs.
107 unfold mkVEnv in *; simpl in *.
108 inversion X33;subst.
109 inversion X35;subst.
110 inversion X37;subst.
111 simpl in *.
112 inversion X38;subst.
113 repeat apply inj_pair2 in H15.
114 subst.
115 inversion X39;subst.
116 inversion X40;subst.
117 inversion H3;subst.
118 clear X40 H3 XF4 H5.
119 inversion X36;subst.
120 inversion X40;subst.
121 inversion X42;subst.
122 simpl in *.
123 inversion X43;subst.
124 repeat apply inj_pair2 in H14.
125 repeat apply inj_pair2 in H16.
126 subst.
```

```
127 unfold xf_SomeCindex at 3 in X43.
128 unfold b_exec,b_eval,b_mod in *.
129 simpl in *.
130 inversion X41;subst.
131 inversion X44;subst.
132 simpl in *.
133 inversion X46;subst.
134 inversion X45;subst.
135 inversion X47;subst.
136 inversion X48;subst.
137 assert (Z : S i = i+1) by omega.
138 rewrite Z; auto.
139 inversion X49.
140 inversion X49.
141 inversion X47.
142 inversion X44.
143 inversion H5.
144 inversion X35;subst.
145 inversion X36.
146 inversion X36.
147 inversion X35.
148 inversion X32.
149 inversion X30.
150 inversion X24.
151 repeat apply inj_pair2 in H6.
152 rewrite H in H6.
153 inversion H6.
154 rewrite H in X21.
155 inversion X21.
156 intros.
157 contradiction.
158 inversion X18.
159 inversion X9.
160 inversion X7.
161 contradiction.
162 Qed.
163
164 (* Lemma succRecW :
165 Proof using Hoare triple rules ... *)
166
167 (* Lemma succRecWByInversion :
168 Proof by inversions ... *)
169
```

```
170 Lemma readPhysicalW (y:Id) table (v:Value)
171         (P' : Value → W → Prop) (fenv: funEnv) (env: valEnv) :
172  {{fun s ⇒ ∃ idxsucc p1, v = cst (option index) (Some idxsucc)
173    ∧ readPhysicalInternal table idxsucc (memory s) = Some p1
174    ∧ P' (cst (option page) (Some p1)) s}}
175 fenv >> (y,v)::env >> ReadPhysical table y {{P'}}.
176 Proof.
177 intros.
178 unfold THoareTriple_Eval.
179 intros; intuition.
180 destruct H.
181 destruct H. intuition.
182 inversion H0;subst.
183 clear k3 t k2 k1 ftenv tenv H1.
184 inversion X;subst.
185 inversion X0;subst.
186 repeat apply inj_pair2 in H7. subst.
187 inversion X2;subst.
188 inversion X3;subst.
189 inversion H0;subst.
190 destruct IdEqDec in H3.
191 inversion H3;subst.
192 clear H3 e X3 H0 XF1.
193 inversion X0;subst.
194 repeat apply inj_pair2 in H7.
195 repeat apply inj_pair2 in H11. subst.
196 inversion X1;subst.
197 inversion X4;subst.
198 repeat apply inj_pair2 in H7.
199 apply inj_pair2 in H9. subst.
200 unfold xf_read at 2 in X4.
201 unfold b_eval,b_exec,b_mod in X4. simpl in *.
202 rewrite H in X4.
203 unfold xf_read,b_eval,b_exec,b_mod in X5. simpl in *.
204 rewrite H in X5.
205 inversion X5;subst. auto.
206 inversion X6.
207 inversion X6.
208 contradiction.
209 Qed.
210
211 (* Other lemmas ... *)
212
```

```
Lemma getFstShadowApplyH' (partition : page) (P : W -> Prop)
                          (fenv: funEnv) (env: valEnv) :
 {{fun s ⇒ P s ∧ partitionDescriptorEntry s ∧
        partition ∈ (getPartitions multiplexer s)}}
  fenv >> env >> (getFstShadowApply partition)
 {{fun sh1 s ⇒ P s ∧ nextEntryIsPP partition sh1idx sh1 s}}.
Proof.
unfold getFstShadowApply.
eapply Apply_VHTT1.
eapply Prms_VHTT1.
eapply Apply_VHTT1.
eapply Prms_VHTT1.
eapply getSh1idxWp.
intros; unfold THoarePrmsTriple_Eval; intros;simpl.
inversion X;subst.
destruct vs; inversion H5.
instantiate (1:= fun vs s => P s /\ partitionDescriptorEntry s /\
     In partition (getPartitions multiplexer s) /\
     vs = [cst index sh1idx]).
intuition. f_equal. auto.
inversion X0. intuition.
destruct vs.
unfold THoareTriple_Eval; intros.
intuition. inversion H3.
destruct vs. Focus 2.
unfold THoareTriple_Eval; intros.
intuition. inversion H3.
unfold mkVEnv. simpl.
eapply weakenEval.
eapply succWp.
simpl; intros.
instantiate (1:= sh1idx).
instantiate (1:= fun s => P s /\ partitionDescriptorEntry s /\
    In partition (getPartitions multiplexer s)).
simpl. intuition.
eapply H in H1.
specialize H1 with sh1idx.
eapply H1. auto.
inversion H3;
intuition. intros; simpl.
unfold THoarePrmsTriple_Eval; intros.
inversion X; subst.
destruct vs; inversion H5.
```

```
256 instantiate (1:= fun vs s => P s /\ partitionDescriptorEntry s /\
257      In partition (getPartitions multiplexer s) /\
258      (exists i : index, succIndexInternal sh1idx = Some i /\
259      vs = [cst (option index) (Some i)])).
260 intuition.
261 destruct H3. exists x. intuition.
262 rewrite H0 in H2. inversion H2; subst.
263 repeat apply inj_pair2 in H6.
264 auto. f_equal. auto.
265 inversion X0. intuition.
266 destruct vs.
267 unfold THoareTriple_Eval; intros.
268 destruct H as [a [b [c d]]].
269 destruct d. destruct H.
270 inversion H0.
271 destruct vs. Focus 2.
272 unfold THoareTriple_Eval; intros.
273 destruct H as [a [b [c d]]].
274 destruct d. destruct H.
275 inversion H0.
276 unfold mkVEnv; simpl.
277 eapply weakenEval.
278 eapply readPhysicalW.
279 simpl; intros. intuition.
280 destruct H3. exists x.
281 unfold partitionDescriptorEntry in H.
282 apply H with partition sh1idx in H1.
283 clear H. intuition. destruct H5.
284 exists x0. intuition.
285 inversion H4;subst. auto.
286 unfold nextEntryIsPP in H5.
287 unfold readPhysicalInternal.
288 rewrite H1 in H5.
289 destruct (lookup partition x (memory s) beqPage beqIndex).
290 destruct v0;try contradiction.
291 unfold cst in H5.
292 apply inj_pairT2 in H5.
293 inversion H5. auto.
294 unfold isVA in H2.
295 destruct (lookup partition sh1idx (memory s) beqPage beqIndex)
296  in H2;try contradiction. auto.
297 Qed.
```

E. Hoare_writeVirtualInv.v

```
(* Imports ... *)

(* Definitions & lemmas ... *)

Lemma writeVirtualInv (vaInCurrentPartition vaChild: vaddr)
       currentPart currentShadow descChild idxDescChild
       ptDescChild ptVaInCurPart idxvaInCurPart vainve
       isnotderiv currentPD ptVaInCurPartpd accessiblesrc
       presentmap ptDescChildpd idxDescChild1 presentDescPhy
       phyDescChild pdChildphy ptVaChildpd idxvaChild
       presentvaChild phyVaChild sh2Childphy ptVaChildsh2 level
                 (fenv: funEnv) (env: valEnv) :
 isnotderiv && accessiblesrc &&
 presentmap && negb presentvaChild = true ->
 negb presentDescPhy = false ->
 {{ fun s : state => propagatedPropertiesAddVaddr
   s vaInCurrentPartition vaChild currentPart currentShadow
   descChild idxDescChild ptDescChild ptVaInCurPart idxvaInCurPart
   vainve isnotderiv currentPD ptVaInCurPartpd accessiblesrc
   presentmap ptDescChildpd idxDescChild1 presentDescPhy
   phyDescChild pdChildphy ptVaChildpd idxvaChild presentvaChild
   phyVaChild sh2Childphy ptVaChildsh2 level }} fenv >> env >>
   WriteVirtual ptVaChildsh2 idxvaChild vaInCurrentPartition
 {{ fun _ s => propagatedPropertiesAddVaddr
   s vaInCurrentPartition vaChild currentPart currentShadow
   descChild idxDescChild ptDescChild ptVaInCurPart idxvaInCurPart
   vainve isnotderiv currentPD ptVaInCurPartpd accessiblesrc
   presentmap ptDescChildpd idxDescChild1 presentDescPhy
   phyDescChild pdChildphy ptVaChildpd idxvaChild presentvaChild
   phyVaChild sh2Childphy ptVaChildsh2 level /\
   readVirtualInternal ptVaChildsh2 idxvaChild s.(memory) =
   Some vaInCurrentPartition }}.
Proof.
intros.
eapply weakenEval. (* weakening precondition *)
eapply writeVirtualWp. (* proven lemma about writeVirtual *)
simpl; intros.
(* The rest is identical to the shallow proof *)
split.
unfold propagatedPropertiesAddVaddr in *.
```

```
41 assert(Hlookup :exists entry,
42  lookup ptVaChildsh2 idxvaChild (memory s) beqPage beqIndex =
43  Some (VA entry)).
44 {assert(Hva: isVA ptVaChildsh2 (getIndexOfAddr vaChild fstLevel) s)
45  by intuition.
46  unfold isVA in *.
47  assert(Hidx:  getIndexOfAddr vaChild fstLevel = idxvaChild)
48   by intuition.
49  clear H. subst.
50  destruct(lookup ptVaChildsh2 (getIndexOfAddr vaChild fstLevel)
51        (memory s) beqPage beqIndex);intros; try now contradict Hva.
52  destruct v; try now contradict Hva.
53  do 2 f_equal.
54  exists v;trivial. }
55  destruct Hlookup as (entry & Hlookup).
56 intuition try assumption.
57 (** partitionsIsolation **)
58 + apply partitionsIsolationUpdateSh2 with entry;trivial.
59 (** kernelDataIsolation **)
60 + apply kernelDataIsolationUpdateSh2 with entry;trivial.
61 (** verticalSharing **)
62 + apply verticalSharingUpdateSh2 with entry;trivial.
63 (** consistency **)
64 + apply consistencyUpdateSh2 with
65     entry vaChild
66     currentPart currentShadow descChild idxDescChild ptDescChild
67     ptVaInCurPart idxvaInCurPart vainve isnotderiv currentPD
68     ptVaInCurPartpd accessiblesrc presentmap ptDescChildpd
69     idxDescChild1 presentDescPhy phyDescChild pdChildphy
70     ptVaChildpd presentvaChild phyVaChild sh2Childphy level;trivial.
71   unfold propagatedPropertiesAddVaddr ;intuition.
72 (** Other Propagated properties **)
73 + rewrite <- nextEntryIsPPUpdateSh2;trivial.
74   exact Hlookup.
75 + apply isVEUpdateSh2 with entry;trivial.
76 + apply getTableAddrRootUpdateSh2 with entry;trivial.
77 + apply entryPDFlagUpdateSh2 with entry;trivial.
78 + apply isVEUpdateSh2 with entry;trivial.
79 + apply getTableAddrRootUpdateSh2 with entry;trivial.
80 + apply isEntryVAUpdateSh2 with entry;trivial.
81 + rewrite <- nextEntryIsPPUpdateSh2;trivial.
82   exact Hlookup.
83 + apply isPEUpdateSh2 with entry;trivial.
```

```
84  + apply getTableAddrRootUpdateSh2 with entry;trivial.
85  + apply entryUserFlagUpdateSh2 with entry;trivial.
86  + apply entryPresentFlagUpdateSh2 with entry;trivial.
87  + apply isPEUpdateSh2 with entry;trivial.
88  + apply getTableAddrRootUpdateSh2 with entry;trivial.
89  + apply entryPresentFlagUpdateSh2 with entry;trivial.
90  + apply isEntryPageUpdateSh2  with entry;trivial.
91  + assert(Hchildren: forall part, getChildren part
92    {| currentPartition := currentPartition s;
93       memory := add ptVaChildsh2 idxvaChild(VA vaInCurrentPartition)
94       (memory s) beqPage beqIndex |} = getChildren part s).
95    { intros; symmetry;
96      apply getChildrenUpdateSh2 with entry;trivial. }
97    rewrite Hchildren in *;trivial.
98  + rewrite <- nextEntryIsPPUpdateSh2;trivial.
99    exact Hlookup.
100 + apply isPEUpdateSh2 with entry;trivial.
101 + apply getTableAddrRootUpdateSh2 with entry;trivial.
102 + apply entryPresentFlagUpdateSh2 with entry;trivial.
103 + apply isEntryPageUpdateSh2  with entry;trivial.
104 + rewrite <- nextEntryIsPPUpdateSh2;trivial.
105   exact Hlookup.
106 + apply isVAUpdateSh2 with entry;trivial.
107 + apply getTableAddrRootUpdateSh2 with entry;trivial.
108 (** new property **)
109 + unfold readVirtualInternal. cbn.
110   assert (Htrue: beqPairs (ptVaChildsh2, idxvaChild)
111       (ptVaChildsh2, idxvaChild) beqPage beqIndex = true).
112   apply beqPairsTrue;split;trivial.
113   rewrite Htrue.
114   trivial.
115 Qed.
```

F. Hoare_initVAddrTable.v

```
(* Imports ... *)

(* Definitions & lemmas ...*)

Lemma succDWp' (x:Id) (v:Value) P (fenv: funEnv) (env: valEnv) :
 ∀ (idx:index),
 {{fun s => P s  /\ idx < tableSize - 1 /\ v=cst index idx}}
  fenv >> (x,v)::env >> SuccD x
 {{fun (idxsuc: Value) (s: state) => P s  /\ idx < tableSize - 1 /\
 v=cst index idx /\ idxsuc=cst (option index (succIndexInternal idx)
  /\ ∃ i, idxsuc = cst (option index) (Some i)}}.
Proof.
intros.
eapply weakenEval.
eapply succDW.
intros.
simpl.
split.
instantiate (1:=idx).
intuition.
intros.
intuition.
destruct idx.
exists (CIndex (i + 1)).
f_equal.
unfold succIndexInternal.
case_eq (lt_dec i tableSize).
intros.
auto.
intros.
contradiction.
Qed.

Lemma initVAddrTableNewProperty table (curidx : index)
                (fenv: funEnv) (env: valEnv) :
 {{ fun s ⇒ (∀ idx : index, idx < curidx →
     (readVirtual table idx (memory s) = Some defaultVAddr) )}}
  fenv >> env >> initVAddrTable table curidx
 {{fun _ s ⇒ ∀ idx, readVirtual table idx s.(memory) =
                     Some defaultVAddr }}.
```

```
41 Proof.
42 unfold initVAddrTable.
43 unfold initVAddrTableAux.
44 assert(H : tableSize + curidx >= tableSize) by omega.
45 revert fenv env H. revert curidx.
46 generalize tableSize at 1 3.
47 induction n.  simpl.
48 (** begin case n=0 *)
49 intros.
50 destruct curidx.
51 simpl in *. omega.
52 (** end *)
53 intros; simpl.
54 eapply Apply_VHTT1.
55 (** begin PS [Val (cst index curidx)] *)
56 instantiate (1:= fun vs s => (forall idx : index,
57 idx<curidx -> readVirtual table idx (memory s) = Some defaultVAddr)
58       /\ vs = [cst index curidx] ).
59 unfold THoarePrmsTriple_Eval.
60 intros.
61 inversion X;subst.
62 destruct vs; inversion H6.
63 destruct vs ; inversion H3 ; subst.
64 intuition.
65 inversion X0;subst.
66 inversion X2.
67 inversion X2.
68 (** end *)
69 intuition; intros; simpl.
70 destruct vs.
71 unfold THoareTriple_Eval;intros.
72 intuition; inversion H2.
73 destruct vs.
74 Focus 2.
75 unfold THoareTriple_Eval;intros.
76 intuition; inversion H2. simpl in *.
77 (*eapply BindMS_VHTT1.*)
78 eapply BindN_VHTT1.
79 (** Begin write Virtual *)
80 unfold THoareTriple_Eval; intros.
81 clear IHn k3 k2 k1 t ftenv tenv env.
82 intuition.
83 inversion H2;subst.
```

```
84  inversion X;subst.
85  inversion X1;subst.
86  inversion X0;subst.
87  inversion X0;subst.
88  repeat apply inj_pair2 in H8. subst.
89  inversion X4;subst.
90  inversion X5;subst.
91  simpl in *.
92  inversion H0;subst.
93  clear X5 H0 XF1.
94  inversion X2;subst.
95  repeat apply inj_pair2 in H8.
96  repeat apply inj_pair2 in H10. subst.
97  unfold b_exec, b_eval, b_mod in *. simpl in *.
98  inversion X3;subst.
99  clear X0 X X1 X2 X3 X4.
100 instantiate (1:= fun s => (forall idx : index,
101 idx<curidx -> readVirtual table idx (memory s) = Some defaultVAddr)
102 /\ v=cst index curidx /\ readVirtual table curidx s.(memory) =
103     Some defaultVAddr).
104 intuition. split.
105 intros.
106 unfold writeVirtualInternal. simpl.
107 unfold readVirtual.
108 unfold add. simpl.
109 assert(Hfalse : Lib.beqPairs (table, curidx) (table, idx)
110         beqPage beqIndex= false).
111     { apply beqPairsFalse. right.
112       apply indexDiffLtb. right;assumption. }
113 rewrite Hfalse.
114 assert (lookup  table idx (Lib.removeDup table curidx (memory n')
115         beqPage beqIndex)
116            beqPage beqIndex = Lib.lookup  table idx  (memory n')
117            beqPage beqIndex) as Hmemory.
118     { apply removeDupIdentity.
119       right.
120       apply indexDiffLtb.
121       left; trivial. }
122 rewrite Hmemory.
123 apply H1 in H0.
124 unfold readVirtual in *. auto. intuition.
125 unfold writeVirtualInternal. simpl.
126 unfold readVirtual.
```

```
127 unfold add. simpl.
128 assert(Htrue : Lib.beqPairs (table, curidx) (table, curidx)
129         beqPage beqIndex= true).
130     { apply beqPairsTrue. intuition. }
131 rewrite Htrue. auto.
132 inversion X5.
133 inversion X5.
134 (** end *)
135 eapply IfTheElse_VHTT1.
136 (** begin LtLtb *)
137 unfold THoareTriple_Eval; intros.
138 clear k3 k2 k1 t tenv ftenv.
139 intuition. subst.
140 inversion X;subst.
141 inversion X0;subst.
142 repeat apply inj_pair2 in H8. subst.
143 inversion X2;subst.
144 inversion X3;subst.
145 simpl in *.
146 inversion H0;subst.
147 clear X3 H0 XF1.
148 inversion X1;subst.
149 inversion X3;subst.
150 repeat apply inj_pair2 in H8.
151 repeat apply inj_pair2 in H10. subst.
152 unfold b_eval, b_exec, xf_Ltb, b_mod in *.
153 simpl in *.
154 inversion X4;subst.
155 instantiate (1:= fun b s => (forall idx : index,
156 idx<curidx -> readVirtual table idx (memory s) = Some defaultVAddr)
157  /\ readVirtual table curidx (memory s) = Some defaultVAddr
158  /\ v=cst index curidx /\ b=cst bool (Index.ltb curidx maxIndex)).
159 intuition.
160 inversion X5.
161 inversion X5.
162 (** end *)
163 simpl.
164 eapply BindS_VHTT1.
165 eapply BindS_VHTT1.
166 (** begin SuccD *)
167 eapply weakenEval.
168 instantiate (2:= fun s => (fun s' => (forall idx : index,
169 idx<curidx -> readVirtual table idx (memory s') = Some defaultVAddr)
```

```
 /\ readVirtual table curidx (memory s) = Some defaultVAddr) s
 /\ curidx < tableSize - 1 /\ v=cst index curidx).
eapply succDWp'.
simpl;intros; intuition.
unfold maxIndex in H4.
inversion H4.
apply inj_pair2 in H5.
symmetry in H5.
apply indexltbTrue in H5.
unfold CIndex in H5.
destruct (lt_dec (tableSize - 1) tableSize).
simpl in *. assumption. contradict n0.
assert (tableSize > tableSizeLowerBound).
apply tableSizeBigEnough.
unfold tableSizeLowerBound in *. omega.
(** end *)
(** begin ExtractIndex *)
intros;simpl.
instantiate (1:= fun v' s => (forall idx : index,
idx<curidx -> readVirtual table idx (memory s) = Some defaultVAddr)
 /\ readVirtual table curidx (memory s) = Some defaultVAddr
 /\ v = cst index curidx /\ curidx < tableSize - 1 /\
 v' = cst index (match succIndexInternal curidx with
                 | Some i => i | None => index_d end) ).
unfold THoareTriple_Eval; intros.
clear k3 k2 k1 t tenv ftenv.
destruct H0.
destruct H1.
destruct H2.
destruct H3.
subst.
destruct H4.
inversion H2.
repeat apply inj_pair2 in H4.
rewrite H4 in *.
inversion X;subst.
inversion X0;subst.
repeat apply inj_pair2 in H10. subst.
inversion X2;subst.
inversion X3;subst.
simpl in *.
inversion H3;subst.
clear X3 H3 XF1.
```

```
213 inversion X1;subst.
214 inversion X3;subst.
215 repeat apply inj_pair2 in H10.
216 repeat apply inj_pair2 in H12. subst.
217 unfold b_eval, b_exec, xf_ExtractIndex, b_mod in *.
218 simpl in *.
219 inversion X4;subst.
220 intuition.
221 inversion X5.
222 inversion X5.
223 (** end *)
224 (* evaluating FVar and Prms*)
225 intros; simpl.
226 eapply QFun_VHTT.
227 econstructor. econstructor.
228 eapply Apply_VHTT2.
229 instantiate(1:=fun vs s => (forall idx : index,
230 idx<curidx -> readVirtual table idx (memory s) = Some defaultVAddr)
231  /\ readVirtual table curidx (memory s) = Some defaultVAddr
232  /\ v = cst index curidx /\ curidx < tableSize - 1
233  /\ v0 = cst index match succIndexInternal curidx with
234      | Some i => i | None => index_d end /\ vs = [v0]).
235 unfold THoarePrmsTriple_Eval;intros.
236 inversion X;subst.
237 destruct vs;inversion H6.
238 inversion X0;subst.
239 inversion X2;subst.
240 inversion X3;subst.
241 inversion X4;subst.
242 inversion H1;subst.
243 inversion X1;subst.
244 destruct vs; inversion H7.
245 inversion X5;subst.
246 inversion X7;subst.
247 inversion X6;subst.
248 destruct vs;inversion H7;subst.
249 destruct vs;inversion H4.
250 intuition.
251 inversion X8;subst.
252 inversion X10.
253 inversion X10.
254 inversion X8.
255 unfold mkVEnv in *; simpl in *.
```

```
intros ;simpl.
destruct vs.
unfold THoareTriple_Eval;intros;intuition.
inversion H6.
destruct vs.
Focus 2.
unfold THoareTriple_Eval;intros;intuition.
inversion H6. simpl in *.
(** recursive call *)
unfold THoareTriple_Eval.
intros. intuition.
inversion H6;subst.
unfold succIndexInternal in *.
destruct curidx.
simpl in *.
case_eq (lt_dec i tableSize);intros; try contradiction.
rewrite H2 in *.
specialize (IHn (CIndex(i+1))).
unfold CIndex in *.
case_eq (lt_dec (i + 1) tableSize);intros.
rewrite H4 in *. simpl in *.
assert (Z : n+(i+1) = S(n+i)) by omega.
rewrite Z in *.
eapply IHn in H as H5.
clear IHn.
eapply H5.
eauto. eauto. eauto. eauto.
clear H6 H5 H k3 k2 k1 t ftenv tenv env idx.
intuition; simpl in *.
assert (Hor: idx={| i:= i; Hi:= Hi |} \/ idx<{| i:= i; Hi:= Hi |}).
    { simpl in *.
      unfold CIndex in H.
      destruct (lt_dec (i + 1) tableSize).
      subst. simpl in *.
      rewrite NPeano.Nat.add_1_r in H.
      apply lt_n_Sm_le in H.
      apply le_lt_or_eq in H.
      destruct H.
      right. assumption.
      left. subst.
      destruct idx. simpl in *. subst.
      assert (Hi = Hi0).
      apply proof_irrelevance.
```

```
299       subst. reflexivity. omega. }
300 destruct Hor.
301 subst. eassumption.
302 apply H1;trivial.
303 assert (i+1<tableSize) by omega;
304 contradiction.
305 (** false case*)
306 revert H. clear;intros.
307 unfold mkVEnv in *; simpl in *.
308 unfold THoareTriple_Eval; intros. intuition.
309 clear k3 k2 k1 ftenv tenv.
310 inversion X;subst.
311 Focus 2. inversion X0.
312 inversion H4.
313 repeat apply inj_pair2 in H3.
314 clear X H4.
315 assert (idx<CIndex (tableSize - 1) \/ idx=CIndex (tableSize - 1)).
316     { destruct idx. simpl in *.
317       unfold CIndex.
318       case_eq (lt_dec (tableSize - 1) tableSize).
319       intros. simpl in *.
320       assert (i <= tableSize -1). omega.
321       apply NPeano.Nat.le_lteq in H4.
322       destruct H4.
323       left. assumption. right. subst.
324       assert (Hi =  Pip_state.CIndex_obligation_1 (tableSize - 1) l).
325       apply proof_irrelevance.
326       subst. reflexivity.
327       intros. omega. }
328 destruct H2.
329 symmetry in H3.
330 apply indexltbFalse in H3.
331 generalize (H1 idx);clear H;intros Hmaxi.
332 apply Hmaxi. subst.
333 apply indexBoundEq in H3.
334 subst. assumption.
335 symmetry in H3.
336 apply indexltbFalse in H3.
337 apply indexBoundEq in H3.
338 subst. assumption.
339 (** end *)
340 Qed.
```